

Spectaculaire spelen voor uw ZX-Spectrum

SPECTACULAR GAMES FOR YOUR ZX SPECTRUM

HAL RENKO / SAM EDWARDS



SPECTACULAIRE SPELEN

voor uw

ZX-SPECTRUM

HAL RENKO

SAM EDWARDS



Addison-Wesley Nederland



Uitgeverij OMIKRON

Omslagontwerp: Stuart Hughes
Illustraties: Agnes Lehár-Graham

Druk: Krips Repro Meppel
Zetwerk: Drukkerij Nico B. de Bruijn B.V., Deventer

ISBN 0-201-16551-1
Addison-Wesley Publishing Company, Inc.
Reading, Massachusetts 01867 U.S.A.

1e druk 1983
2e druk 1984

© Uitgeverij OMIKRON

Niets uit deze uitgave mag worden vermenigvuldigd of openbaar worden gemaakt door middel van druk, fotokopie of op welke andere wijze ook, zonder voorafgaande schriftelijke toestemming van Uitgeverij OMIKRON

Ondanks alle aan de samenstelling van dit boek bestede zorg, kan noch de redactie noch de uitgever aansprakelijkheid aanvaarden voor eventuele schade, die zou kunnen voortvloeien uit enige fout, die in deze uitgave zou kunnen voorkomen.

INHOUD

Zombies in het moeras	10
Galactische monsters	14
Keyboard Memory	18
Las Vegas a GoGo *	20
Papapapegaai	28
Duindigt	30
Het niet zo verschrikkelijke kleurenprobleem	35
Qui Vive *	38
STM	48
Van 5 naar 0	51
Escher	54
Het rekenwonder	58
De Shakespeare shuffle	61
Explosie	65
New York, New York	71
Code kraken	76
Blackbox	79
Schatzoeken	83
I.T., De adventure van de eeuw *	87
De wolf en de 5 kleine geitjes	99
Stuiterbal	103
Formule 1-race	104
De marktkramer	107
Doorzakken	109
UFO aanval	112
Mini Mancala	115
Stop it!	120
BAAB	122
Klinkers en medeklinkers	125
Maanfase astrologie	128

* Deze spelen vereisen een ZX-Spectrum met 48 Kb RAM

Dit boek dankt zijn ontstaan aan een bezoek dat wij aan de Technische Hogeschool Twente brachten. Hier ontmoetten we de volgende spelletjes fanaten:

Hermie Hermens, Peter de Vries, Arie Jan Stapel, Arend Rensink, Mark Sijbenga, Erik Groenhuis, Wim Koolhoven, Ronald Bosch, Berry Hoogstraten, Claudia Dispa, Alex Pelsmaeker en Kasper Boon.

Wij danken al deze computer freaks voor de vele uren die zij samen met ons hebben besteed aan het tot stand komen van deze spelen.



VOORWOORD

Iedereen weet natuurlijk dat computers voor allerlei doeleinden gebruikt worden, zoals voor financiële berekeningen, zakelijke toepassingen en tekstverwerking.

Toch kunnen ze ons ook op een heel andere wijze van nut zijn. We moeten hierbij onwillekeurig aan het spreekwoord „na gedane arbeid is het goed rusten” denken. Ontspanning is inderdaad een noodzakelijk deel van ons leven. Ter ontspanning lezen we boeken, hebben we onze hobby's en vermaken we ons met spelletjes.

Brueghels beroemde schilderij „Spelende kinderen” bewijst op treffende wijze dat dit in feite een eeuwenoude waarheid is.

Dit boek vol spelen is dan ook voor die broodnodige ontspanning bedoeld en wel voor iedereen van 6 tot 96 jaar. Er zijn spelen om in je eentje te spelen en andere om met het gezin of vrienden te spelen.


Er zijn aktiespelen bij om Uw snelheid en reflexen te testen, puzzels en ware hersenkrakers om Uw geest te scherpen en bordspelen waarin we onze krachten met die van de computer kunnen meten.

Natuurlijk zijn er ook enkele arcade-achtige spelen.

En tenslotte ... wilt U een avontuur beleven ... I.T., de adventure van de eeuw brengt U in een geheel andere wereld.

Een klasse apart wordt gevormd door de spelen die we snel kunnen intypen maar die toch voor zeer intrigerende effecten zorgen.


Als we weer naar het schilderij kijken komt automatisch een gedachte bij ons op ... als Brueghel nu leefde, zou zijn schilderij er dan zo uitzien?



ZOMBIES IN HET MOERAS

Pas op! Zombies dwalen rond en toch zult U het moeras moeten oversteken. Om deze opwindende taak te kunnen volbrengen moet U gebruik maken van vloten.

Aan het begin van het spel ziet het moeras er als volgt uit:



U staat in de rechter benedenhoek en U moet trachten de schuin hier tegenoverliggende hoek te bereiken om zo uiteindelijk te ontsnappen.

Gebruik de cursorbesturingstoetsen om een vlot naast hetgene waarop U staat uit te werpen. U moet eveneens deze toetsen gebruiken om er vervolgens op te gaan staan.

Misschien lijkt dit alles nog niet zo moeilijk maar pas op! Van tijd tot tijd komt één van de Zombies uit het moeras en hij dwaalt hierna over de vloten die U heeft uitgeworpen. Als hij U op zijn weg tegenkomt is het natuurlijk met U gedaan... Gelukkig zien Zombies niet zo erg goed en hierdoor vallen ze toch nog wel eens in het moeras. Om dit spannende spel te kunnen winnen moet U een aparte strategie ontwikkelen en dit nu maakt het spel zo aantrekkelijk. Hoe langer en ingewikkelder Uw weg is, hoe moeilijker de Zombies U te pakken kunnen krijgen, maar tevens

hoe langer het duurt voordat U aan de overkant bent aangeland. Dus
 ... hoe meer Zombies er zullen verschijnen ...
 Kortom ... horror op de buis!

```

      5 REM ZOMBIES IN HET MOERAS
      10 DIM a(9,9)
      20 FOR c=1 TO 9
      30 LET a(1,c)=-1: LET a(c,1)=-
1 40 LET a(9,c)=-1: LET a(c,9)=-
1 50 NEXT c
      60 LET a(2,1)=1: LET a(1,2)=1
      70 LET e$=""
      80 LET p$=""
      90 LET m$=""
     100 LET d$=""
     110 LET b$=""
     120 GO TO 3000
1000 REM print t$ op x,y
1010 PRINT AT 2*y,3*x; PAPER 5;t
$( TO 3)
1020 PRINT AT 2*y+1,3*x; PAPER 5
;t$(4 TO )
1030 RETURN
1100 REM initialisatie
1110 LET m=0: LET px=0: LET py=0
: LET a(6,6)=1
1120 FOR k=4 TO 17: PRINT AT k,6
;b$: NEXT k
1130 LET x=px: LET y=py: LET t$=
p$: GO SUB 1000
1140 RETURN
1200 REM verplaatsen speler
1210 REM instructie invoeren
1220 LET t$=INKEY$
1230 LET dx=(t$="8")-(t$="5")
1240 LET dy=(t$="6")-(t$="7")
1250 IF (dx+dy)=0 THEN GO TO 122
0
1260 GO TO 1300+100*a(px+dx,py+d
y)
1300 REM vlot gooien
1310 LET x=px+dx: LET y=py+dy: L
ET a(x,y)=1
1320 LET t$=e$
1330 GO SUB 1000
1340 RETURN
1400 REM verplaatsen
1410 LET x=px: LET y=py: LET t$=
e$: GO SUB 1000
1420 LET px=px+dx: LET py=py+dy

```

```

1430 IF (px=1) OR (py=1) THEN GO
    TO 1500
1440 LET x=px: LET y=py: LET t$=
p$: GO SUB 1000
1450 RETURN
1500 REM gewonnen
1510 PRINT AT 20,0;" GEFELICITEE
RD"
1520 LET s$=INKEY$
1530 IF s$="" THEN GO TO 1500
1540 RUN
1600 REM monster
1610 LET mx=7: LET my=7
1620 LET x=mx: LET y=my: LET t$=
m$: GO SUB 1000
1630 GO SUB 2100+10*INT (RND*2)
1640 LET lx=dx: LET ly=dy
1700 REM nieuwe richting kiezen
1710 IF a(mx+lx,my+ly)<1 THEN GO
    TO 1900
1720 IF a(mx+ly,my-lx)=1 OR a(mx
-ly,my+lx)=1 THEN GO TO 2000
1800 REM zelfde richting
1810 LET dx=lx: LET dy=ly: GO TO
    2200
1900 REM moeras in
1910 GO SUB 2100+10*INT (RND*4)
1920 IF a(mx+dx,my+dy)=-1 THEN G
    O TO 1910
1930 GO TO 2200
2000 REM richting bepalen
2010 GO SUB 2100+10*INT (RND*4)
2020 IF (dx=-lx AND dy=-ly) OR (
a(mx+dx,my+dy)<1) THEN GO TO 180
0

```



```

2030 GO TO 2200
2100 LET dx=0: LET dy=-1: RETURN
2110 LET dx=-1: LET dy=0: RETURN
2120 LET dx=0: LET dy=1: RETURN
2130 LET dx=1: LET dy=0: RETURN
2200 REM verplaatsen
2210 LET x=mx: LET y=my: LET t$=
e$: GO SUB 1000
2220 LET mx=mx+dx: LET my=my+dy
2230 LET lx=dx: LET ly=dy
2240 IF a(mx,my)=0 THEN GO TO 23
00
2250 LET x=mx: LET y=my: LET t$=
m$: GO SUB 1000
2260 IF mx=px AND my=py THEN GO
TO 2400
2270 GO TO 1700
2300 REM monster verdronken
2310 LET a(mx,my)=-1
2320 LET x=mx: LET y=my: LET t$=
d$: GO SUB 1000
2330 RETURN
2400 REM speler gepakt
2410 PRINT AT 20,0;"JAMMER, U BE
NT OPGEGETEN"
2420 LET s$=INKEY$
2430 IF s$="" THEN GO TO 2420
2440 RUN
3000 REM hoofdprogramma
3010 GO SUB 1100: REM initialisa
tie
3020 GO SUB 1200: REM verplaatse
n speler
3030 LET m=m+1
3040 IF (px+py)*RND>4 OR m<5 THE
N GO TO 3020
3050 GO SUB 1600: REM monster
3060 GO TO 3020

```


GALACTISCHE MONSTERS

Eindelijk is het zover, op het x-y-Gz radarscherm is duidelijk zichtbaar dat we de Melkweg naderen. Nog slechts 2.56 protoseconden en we zijn weer veilig in ons eigen melkwegstelsel!

Maar er dreigt gevaar en diep in Uw hart weet U heel goed wat het is. Uw blik is strak gericht op het LC2F scherm. Nog geen alarm ... De spanning wordt ondraaglijk ... 2,5 protoseconden ... 2,0 protoseconden ... 1,5 protoseconden ... oh nee! Daar is het ...

DE VAN ALLEN SQUARE STONE BELT

Er zijn veel van deze beruchte velden, velden waarin de angstaanjagende galactische monsters loeren. Geen wapen kan iets uitrichten tegen deze wraakzuchtige monsters; het enige wat U doen kunt is ze trachten te ontwijken.

Als het spel begint ziet U de VAN ALLEN SQUARE STONE BELT en dan kunnen we onze bewegingen sturen aan de hand van de volgende toetsen.

- ↑ één vierkant naar boven
- ← één vierkant naar links
- één vierkant naar rechts
- ↓ één vierkant naar beneden

Als U de VAN ALLEN SQUARE STONE BELT een tweede maal oversteekt zullen de monsters aggressiever worden.

Op de een of andere manier hebben ze een voorgevoel van iedere beweging. Als ze er in slagen U vijf keer met succes aan te vallen, dan is het definitief met U gedaan...

Succes ... slechts een paar protoseconden te gaan en er wacht een lekker bakkie koffie!

```


10 REM GALACTISCHE MONSTERS
90 GO SUB 300
100 FOR t=1 TO 11
110 FOR z=1 TO 11
120 PRINT AT 5+t,9+z;"0"
130 NEXT z
140 NEXT t
150 IF k>4 THEN GO TO 700
160 LET scor=scor+num-turn-5*k
170 LET num=num+20
180 LET turn=0
190 BEEP .5,3
200 FOR t=1 TO 11
210 PRINT AT 13,9+t;"P"
220 NEXT t
230 PRINT AT 6,15;"R"
240 LET x1=6: LET y1=15
250 FOR t=1 TO 11
260 LET x(t)=13: LET y(t)=9+t
270 NEXT t
280 LET chance=chance+.1
290 BEEP .2,1
300 IF k>4 THEN GO TO 700
310 PRINT AT x1,y1;"0"
320 LET turn=turn+1
330 LET t$=INKEY$
350 LET x1=x1+(t$="6")-((t$="7"
) AND (x1>6))
360 LET y1=y1+((t$="8") AND (y1
<20))-((t$="5") AND (y1>10))
370 PRINT AT x1,y1;"R"
380 FOR t=1 TO 11
390 IF ((x1=x(t)) AND (y1=y(t))
) THEN BEEP .5,1: BEEP .5,2: LET
k=k+1
410 NEXT t
420 IF x1=16 THEN GO TO 100
430 GO SUB 500
440 GO TO 300
500 FOR t=1 TO 11
510 PRINT AT x(t),y(t);"0"
520 IF RND<chance THEN GO TO 59
0
530 LET a=INT (RND*4): GO TO 54
0+a*10
540 LET x(t)=x(t)+(x(t)<16): GO
TO 620
550 LET x(t)=x(t)-(x(t)>6): GO
TO 620
560 LET y(t)=y(t)+(y(t)<20): GO
TO 620
570 LET y(t)=y(t)-(y(t)>10): GO
TO 620
580 GO TO 620
590 LET x2=x1-x(t): LET y2=y1-y
(t)
600 IF RND>.8 THEN LET x(t)=x(t
)+SGN (x2): GO TO 620

```

```

610 LET y(t)=y(t)+SGN (y2)
620 PRINT AT x(t),y(t); "P"
630 NEXT t
640 FOR u=1 TO 11
650 PRINT AT x(u),y(u); "P"
660 NEXT u
670 RETURN
700 PRINT "HET EINDE VAN DIT SP
ELLETJE"

```



```

710 PRINT "UW SCORE IS ";scor
720 STOP
800 REM initialisatie
810 DIM x(12): DIM y(12)
820 LET chance=0: LET scor=0
830 LET turn=0: LET k=0
840 LET num=0
850 FOR i=0 TO 7: READ fig: POK
E USR "p"+i,fig: NEXT i

```

```

860 FOR i=0 TO 7: READ fig: POK
E USR "q"+i,fig: NEXT i
870 FOR i=0 TO 7: READ fig: POK
E USR "r"+i,fig: NEXT i
900 RETURN
1000 DATA BIN 00000000,BIN 01100
110,BIN 01000010,BIN 00011000,BI
N 00011000,BIN 01000010,BIN 0011
1100,0
1010 DATA 0,BIN 01111110,BIN 011
11110,BIN 01111110,BIN 01111110,
BIN 01111110,BIN 01111110,0
1020 DATA BIN 00011000,BIN 00011
000,BIN 00111100,BIN 01011010,BI
N 10011001,BIN 00111100,BIN 0010
0100,BIN 01100110
9999 FOR i=1 TO 100: PRINT AT 10
,10;"0";: NEXT i

```

KEYBOARD MEMORY

Bij dit spel gebruiken we het toetsenbord (Engels: keyboard) van de computer om Uw geheugen te testen. Achttien toetsen verbergen n.l. een symbool en wel op dezelfde manier waarop een speelkaart die op z'n kop ligt, zijn waarde verbergt.

Er zijn maar negen verschillende symbolen. Dit houdt in dat ieder symbool onder precies twee toetsen voorkomt. De toetsen waarom het hier gaat worden steeds op het scherm getoond. Als we nu een toets indrukken zal de letter die door deze toets verborgen wordt op het scherm worden getoond. Als U nu twee toetsen na elkaar indrukt die de zelfde letter verbergen dan zullen deze toetsen van het scherm verdwijnen.

Q	W	E	R	T	Y
A	S	D	F	G	H
Z	X	C	V	B	N

Uw eerste pogingen zijn natuurlijk gissingen maar door de posities van de getoonde karakters te onthouden zult U spoedig ontdekken welke toetsen paren vormen.

Hoeveel beurten heeft U nodig om het scherm leeg te krijgen?


```
10 REM KEYBOARD MEMORY
20 DIM t$(1,18): DIM w$(1,18):
DIM a$(1,2): DIM n(2)
30 LET r=INT (RND*9+1)
40 CLS : RANDOMIZE
50 LET tu=0: LET te=0
60 LET a$="qwertyasdfghzxcvbn"
70 LET b$="175662951734409032"
80 FOR n=1 TO 18
```



```

90 LET t$(1,n)=a$(n)
100 LET dis=n+r
110 IF dis<19 THEN GO TO 130
120 LET dis=dis-18
130 LET w$(1,n)=b$(dis)
140 NEXT n
150 LET n(1)=0: LET n(2)=0
160 FOR k=1 TO 3
170 FOR j=1 TO 6
180 PRINT AT 4*k,4*j;t$(1,(k-1)
#6+j)
190 NEXT j
200 NEXT k
210 FOR k=1 TO 2
220 LET k$=INKEY$
230 IF k$="" THEN GO TO 220
240 LET q$(1,k)=k$
250 LET st=0
260 LET tu=tu+1
270 FOR n=1 TO 18
280 LET kk=INT ((n-0.5)/6+1)
290 LET j=n-(kk-1)*6
300 IF t$(1,n)<>q$(1,k) THEN GO
TO 330
310 PRINT AT 4*kk,4*j;w$(1,n)
320 LET n(k)=n
330 NEXT n
340 IF n(1)=n(2) THEN BEEP .5,1
2: GO TO 220
350 NEXT k
360 IF w$(1,n(1))<>w$(1,n(2)) T
HEN GO TO 150
370 LET te=te+1
380 LET t$(1,n(1))=""
390 LET t$(1,n(2))=""
400 IF te<9 THEN GO TO 150
410 CLS
420 PRINT AT 15,5;"Aantal beurt
en=" ;tu

```



LAS VEGAS A GOGO

Heeft U zich wel eens verbaasd over de vasthoudendheid waarmee sommige mensen met gokautomaten, zoals een fruitautomaat, hun geluk beproeven? En heeft U wel eens het verbeterd gezicht gezien van de persoon die al zijn geld vergokte? Welnu, U zult gauw genoeg merken waarom dat gokken toch zo fascinerend is! Als U dit programma laadt wordt Uw computer plotseling in een geweldige fruitautomaat omgetoerd. Alle spelinstructies worden op het scherm getoond. Soms kunt U uit een aantal mogelijkheden kiezen, zoals:

INSERT, HOLD, PLAY OR END


U kunt nu de keuze aan de computer kenbaar maken, steeds door de eerste letter van het door U gekozen alternatief in te typen. Bijvoorbeeld P zal de wielen van het geluk opnieuw in beweging zetten Deze wielen zijn van links naar rechts met 1, 2 en 3 genummerd. Als U een of meer symbolen in het venster van de automaat wilt vasthouden, geef dan na de H (Hold) ingetypt te hebben de nummers van Uw keuze op. Daar gaan we dan ... de wielen draaien, het apparaat toont zijn bekende flitsende betovering en fluistert ons toe ... win de Jackpot!

```
10 REM ***LAS VEGAS A GOGO***
100 DIM r(3,13): DIM j(13): LET
j(5)=1: LET j(6)=1: LET j(7)=1
110 LET w$="AKQJ7*$0/=#%"
120 FOR i=5 TO 12: READ r(2,i),
r(3,i): NEXT i
130 DATA 8,40,8,40,7,30,5,25,3,
20,3,20,2,10,2,10
140 DIM h(3): DIM p(3): DIM c(4)
): DIM d(4)
150 DIM k(3): DIM v(4): DIM o(3)
): DIM a(3)
160 DIM f(3): DIM g(4): DIM e(4)
,2)
170 FOR i=1 TO 3: READ p(i): NE
XT i
180 FOR i=1 TO 4: READ d(i): RE
AD e(i,1): READ e(i,2): NEXT i
190 DATA 12,14,16,12,1,14,13,2,
20,15,3,14,16,2,9
```

```

200 LET b$="": REM
13*
210 LET nt=-1: LET nj=4: LET ni
=0: LET wi=0
220 LET hb=0: LET do=0: LET dc=
0: LET ja=0
230 LET jc=0: LET mo=0: LET m# =
0: LET r=0: LET gw=0: LET nh=0:
LET ng=0
240 FOR i=1 TO 3: LET v(i)=INT
(RND*14): NEXT i
600 GO TO 4000
1000 REM ***$$. KNIPPEREN EN INKE
Y$ INLEZEN***
1010 PRINT AT r,c; OVER 1; FLASH
1;= $;
1020 LET i$=INKEY$: IF i$="" THE
N GO TO 1020
1030 PRINT AT r,c; PAPER c;t$;
1040 RETURN
1100 REM ***GELD INWERPEN***
1110 FOR i=SGN (in) TO in STEP 5
GN (in)
1120 PRINT AT 0,8;mo+i;" "; BEE
P .1,40: NEXT i
1130 LET mo=mo+in: RETURN
1200 REM ***STOP DUBBELE PUNTEN*
**
1210 LET do=0: PRINT AT 6,11; PA
PER 2;" ";
1220 FOR i=1 TO 4: PRINT AT 7,d(
i);w$(i);: LET c(i)=0: NEXT i: R
ETURN
1300 REM ***VEEG LAATSTE REGEL S
CHOON***
1310 PRINT AT 21,0;TAB 31;"": R
ETURN
1400 REM ***JACKPOT***
1410 FOR t=1 TO 4: FOR c=1 TO jc
: PRINT AT 0,11+c;" ";: NEXT c
1420 FOR c=1 TO jc: BEEP .1,10+2
*c: PRINT AT 0,11+c;"J";: NEXT c
1430 NEXT t: RETURN
1500 REM ***LAAT HOLD KNIPPEREN
EN LEES INKEYS$***
1510 FOR i=1 TO 3: IF h(i)=0 THE
N PRINT AT 11,p(i); FLASH 1; PAP
ER 2; INK 7; OVER 1;"H";
1520 NEXT i: LET ho=0
1530 LET i$=INKEY$: IF i$="" THE
N GO TO 1530
1540 IF i$<>"1" AND i$<>"2" AND
i$<>"3" THEN GO TO 1570
1550 LET i=VAL i$: PRINT AT 11,p
(i); PAPER 2; INK 7;"H";
1560 LET ho=1: LET h(i)=1: GO TO
1530
1570 FOR i=1 TO 3: IF h(i)=0 THE
N PRINT AT 11,p(i); PAPER 2;" ";

```



```


1530 NEXT i: LET nh=nh+h0: RETURN
2000 REM ***SCHERMLAYOUT***
2010 PRINT AT 5,11; PAPER 3;"
2020 FOR i=0 TO 4: PRINT AT 6+i,
10; PAPER 3;" "; PAPER 2;"
"; PAPER 3;" "; PAPER 7;" "; PA
PER 3;" ";: NEXT i
2030 PRINT AT 6,20; PAPER 2;" ";
: PRINT AT 10,19; PAPER 3;" ";
2040 FOR i=0 TO 2: PRINT AT 9,12
+2*i; PAPER 6;" ";: NEXT i
2050 PRINT AT 11,11; PAPER 5;"
2060 FOR i=0 TO 2: PRINT AT 11,1
2+2*i; PAPER 2;" ";: NEXT i
2070 FOR i=0 TO 3: PRINT AT 12+i
,12; PAPER 5;" ";: NEXT i
2080 FOR i=0 TO 3: FOR k=0 TO 4+
2*i: PRINT AT 16+i,12-i+k; PAPER
1;" ";: NEXT k: NEXT i
2090 PRINT AT 0,0;"TOTAL: 0";
2100 PRINT AT 0,0;"--%--= 1";
2110 FOR i=12 TO 5 STEP -1
2120 PRINT AT 21-i,0;"-";w$(i);w
$(i);"- = ";r(2,i);
2130 PRINT AT 21-i,22;w$(i);w$(i
);w$(i);" = ";r(3,i);
2140 IF j(i) THEN PRINT "+J";
2150 NEXT i: RETURN
2300 REM ***VARIABLEN AANPASSEN
***
2310 LET nt=nt+1: IF wi>0 THEN L
ET hb=1: LET wi=0
2320 FOR i=1 TO 3: LET h(i)=0: P
RINT AT 11,p(i); PAPER 2;" ";: N
EXT i
2330 IF d0 THEN GO TO 2360
2340 FOR i=1 TO 4: IF c(i)=0 THE
N GO TO 2370
2350 LET c(i)=c(i)-1: IF c(i)>0
THEN GO TO 2370
2360 LET nj=nj+1: PRINT AT 7,d(i
);w$(i);

```

```

2370 NEXT i: GO TO 2390
2380 LET dc=dc-1: IF dc=0 THEN G
O SUB 1200
2390 IF ja THEN PRINT AT 0,11+jc
;" "; LET jc=jc-1: IF jc=0 THEN
LET ja=0
2400 IF mo>mm THEN LET mm=mo
2410 RETURN
2500 REM ***NIET GENOEG GELD***
2510 LET hb=1: GO SUB 1200
2520 LET ja=0: PRINT AT 0,10;TAB
31;" ";
2530 GO SUB 1300: PRINT AT 21,8;
PAPER 5;"INSERT OR END";
2540 LET s$="INSERT (I)": LET t$
=b$: LET cl=7: LET r=2: LET c=0:
GO SUB 1000
2550 IF i$="P" OR i$="p" THEN GO
TO 2540
2560 RETURN
2600 REM ***HOLD MOGELIJK***
2610 GO SUB 1300: PRINT AT 21,2;
PAPER 5;"INSERT, HOLD, PLAY OR
END";
2620 GO SUB 1500: RETURN

```



```

2700 REM ***GEEN HOLD MOGELIJK**
*
2710 GO SUB 1300: PRINT AT 21,5;
PAPER 5;"INSERT, PLAY OR END";
2720 LET s$="???": LET t$=" ";
LET cl=1: LET r=18: LET c=13: G
O SUB 1000
2730 RETURN
2800 REM ***GOKKEN OF WINNEN***
2810 GO SUB 1300
2820 IF hb THEN PRINT AT 21,7;" "
; GO TO 2840
2830 PRINT AT 21,4; PAPER 5;"HOL
D, ";
2840 PRINT ; PAPER 5;"GAMBLE OR
COLLECT";
2850 LET s$="???": LET t$=" ";
LET cl=1: LET r=18: LET c=13: G
O SUB 1000
2860 RETURN
2900 REM ***PLAY***
2910 FOR i=1 TO 3: IF h(i) THEN
GO TO 2940

```



```

2920 LET o(i)=v(i): LET fi=INT (
  RND*100)+1: LET v(i)=fi: IF fi<5
  THEN GO TO 2940
2930 LET v(i)=5+(fi>7)+(fi>10)+(
  fi>13)+(fi>36)+(fi>49)+(fi>68)+(
  fi>87)
2940 NEXT i
2950 FOR c=6 TO 9: PRINT AT c,20
  " ";: PRINT AT c+1,20; PAPER 2;
  " ";
2960 FOR i=1 TO 15: NEXT i: NEXT
  c
2970 FOR c=10 TO 7 STEP -1: PRIN
  T AT c,20; PAPER 3;" ";: PRINT A
  T c-1,20; PAPER 2;" ";
2980 FOR i=1 TO 15: NEXT i: NEXT
  c
2990 FOR i=1 TO 3: LET f(i)=0: N
  EXT i
3000 FOR i=0 TO 70: FOR j=1 TO 3
3010 IF h(j)=1 THEN GO TO 3060
3020 IF f(j)=1 THEN GO TO 3050
3030 LET o(j)=o(j)+1: IF o(j)>13
  THEN LET o(j)=1
3040 IF (i>10+13*j) AND o(j)=v(j
  ) THEN LET f(j)=1: BEEP .02,8*j
3050 PRINT AT 9,p(j); PAPER 6;w$
  (o(j));
3060 NEXT j: NEXT i: RETURN
3100 REM ***JOKERS***
3110 LET jo=0: FOR i=1 TO 3: LET
  v=v(i)
3120 IF v>4 THEN LET fv=v: GO TO
  3150
3130 LET jw=i: LET jo=jo+1: IF c
  (v)>0 THEN GO TO 3150
3140 PRINT AT 7,d(v); PAPER 3;"
  ";: LET c(v)=20: LET nj=nj-1
3150 NEXT i: LET dc=(nj=0): IF d
  o THEN LET dc=15: PRINT AT 8,11;
  PAPER 6;"DOUBLE!";
3160 RETURN
3200 REM ***BEREKEN WINST***
3210 LET hv=5: GO TO (3220 AND j
  o=0)+(3250 AND jo=1)+(3220 AND j
  o=2)+(3270 AND jo=3)
3220 LET hv=fv
3230 GO TO 3270
3250 LET v(4)=15: LET jr=jw+1: L
  ET jl=jw-1: IF jl=0 THEN LET jl=
  4
3260 LET hv=v(jr): IF v(jr)>v(jl
  ) THEN LET hv=v(jl)
3270 FOR i=1 TO 3: LET k(i)=v(i)
  : IF v(i)<5 THEN LET k(i)=hv
3280 NEXT i: IF k(1)<>k(2) OR k(
  2)<>k(3) THEN GO TO 3300
3290 IF j(k(1)) THEN LET ja=1: L
  ET dc=15: GO SUB 1400

```

```


3300 LET w1=0: LET w2=0
3310 FOR i=1 TO 3: IF k(i)=13 TH
EN LET w1=w1+1
3320 NEXT i: LET ns=1+(k(1)=k(2)
)+(k(2)=k(3)): LET w2=r(ns,k(2))
3330 LET wi=w2: IF w1>wi THEN LE
T wi=w1
3340 IF ja AND wi>0 AND wi<10 TH
EN LET wi=10
3350 IF do THEN LET wi=2*wi
3360 RETURN
3400 REM ***GOKKEN***
3410 LET dt=1: LET g(1)=2*wi: LE
T g(2)=0: LET g(3)=INT (3*wi/2):
LET g(4)=INT (wi/2)
3420 LET ng=ng+1: GO SUB 1300: P
RINT AT 21,12: PAPER 5;"STOP";
3430 FOR i=1 TO 3: PRINT AT i,8;
PAPER 6;";": NEX
T i: REM 15*
3440 LET r=r+1: IF r>4 THEN LET
r=1
3450 PRINT AT e(r,1),e(r,2); PAP
ER 6;g(r);: BEEP .003,10+3*r
3460 IF i$<>"S" AND i$<>"s" THEN
LET i$=INKEY$: GO TO 3480
3470 LET dt=(2+RND)*0.5*dt+1.3:
IF dt>250 THEN LET dt=250
3475 FOR i=1 TO dt: NEXT i: IF d
t>200 THEN GO TO 3490
3480 PRINT AT e(r,1),e(r,2); PAP
ER 6;";": GO TO 3440
3490 FOR i=1 TO 3: PRINT AT i,8;
TAB 31;"";: NEXT i
3500 LET gw=gw+g(r)-wi: LET wi=g
(r): RETURN
3600 REM ***EINDE VAN HET SPEL**
*
3610 CLS : PRINT : PRINT "BEDRAG
EN": PRINT
3620 PRINT " INGEWORPEN:";TAB 18
;"FL";: LET x=ni: GO SUB 3800
3630 PRINT " UITBETAALD:";TAB 18
;"FL";: LET x=mo/4: GO SUB 3800
3640 PRINT " MAX IN 1 BEURT: FL
";: LET x=mm/4: GO SUB 3800
3650 IF gw>=0 THEN PRINT " GAMBL
E WINST: FL";: LET x=gw/4: GO
SUB 3800
3660 IF gw<0 THEN PRINT " GAMBLE
VERLIES: FL";: LET x=-gw/4: G
O SUB 3800
3670 PRINT : PRINT : PRINT "AANT
AL KEREN HOLD:";TAB 20;: LET x=n
h: GO SUB 3900
3680 PRINT "AANTAL KEREN GAMBLE:
";TAB 20;: LET x=ng: GO SUB 3900
3690 PRINT "AANTAL BEURTEN:
";TAB 20;: LET x=nt: GO SUB 3900

```

```

3700 RETURN
3800 LET l=LEN STR$ X-LEN STR$ I
NT X
3810 LET X$=STR$ X+("0" AND l=2)
+(".00" AND l=0)
3820 FOR i=LEN X$ TO 5: PRINT "
";: NEXT i: PRINT X$: RETURN
3900 FOR i=LEN STR$ X TO 2: PRIN
T " ";: NEXT i: PRINT X: RETURN
4000 REM ***HOOFDPROGRAMMA***
4010 GO SUB 2000: REM SCHERM
4020 GO SUB 2300: REM VARIABELEN
4030 LET ca=2*hb+(mo>1)+1: GO SU
B (2500 AND ca=1)+(2600 AND ca=2
)+(2500 AND ca=3)+(2700 AND ca=4
)
4040 PRINT AT 3,8;TAB 31;"";: LE
T ca=(i$="I" OR i$="i")+2*(i$="P
" OR i$="p")+3*(i$="E" OR i$="e"
)
4050 GO TO (4030 AND ca=0)+(4100
AND ca=1)+(4200 AND ca=2)+(4700
AND ca=3)
4100 LET ni=ni+1: LET in=4: GO S
UB 1100: IF mo>mm THEN LET mm=mo
4110 GO TO 4030
4200 LET in=-2: GO SUB 1100
4210 GO SUB 2900: REM PLAY
4230 GO SUB 3100: REM JOKERS
4240 GO SUB 3200: REM WINST

```



```

4250 IF h(1)+h(2)+h(3)>0 THEN GO
TO 4270
4260 LET hb=0: LET lw=wi: GO TO
(4290 AND wi>0)+(4020 AND wi<=0)
4270 LET hb=1: IF wi>lw THEN GO
TO 4290
4275 IF lw=0 THEN GO TO 4020
4280 PRINT AT 3,10;TAB 31;"";: P
RINT AT 3,11: PAPER 6;"VERLOREN"
;: BEEP 2,10: GO TO 4020
4290 PRINT AT 3,10: PAPER 6;"GEW
ONNEN ";wi;: FOR i=1 TO wi: BEEP
.1,20: NEXT i
4300 IF mo<2 THEN LET hb=1

```

```

4310 GO SUB 2800: REM INSTRUCTIE
      INLEZEN
4320 LET ca=((i$="H" OR i$="h")
AND NOT hb)+2*(i$="G" OR i$="g")
+3*(i$="C" OR i$="c")
4330 GO TO (4310 AND ca=0)+(4600
      AND ca=1)+(4500 AND ca=2)+(4400
      AND ca=3)
4400 LET in=wi: GO SUB 1100: GO
      TO 4020
4500 LET hb=1: GO SUB 3400: REM
      GOKKEN
4510 IF wi>0 THEN GO TO 4290
4520 GO TO 4280
4600 LET wi=0: GO TO 4020
4700 GO SUB 3600: REM EINDE

```

PAPAPAPEGAAI

Papapapegaai is een erg eenvoudig competitie spel. Leuk om op een regenachtige dag te spelen met een vriend (vooral als die vriend iets gedronken heeft!).

Aan het begin is de computer zo onbeleefd om U te vragen:

BEN JE KLAAR PAPAPAPEGAAI?

Nota Bene, hij noemt U papapapegaai! Als U een willekeurige toets indrukt laat hij onmiddellijk een letter zien. U moet diezelfde letter intoetsen en wel zo snel U kunt. Daar U iedere letter opnieuw moet intypen bent U werkelijk een papegaai! Het spel wordt automatisch beëindigd na ca. 10 seconden en de computer toont Uw score.

Hoe goed bent U als papegaai? Ga nu een wedstrijd met Uw vrienden aan en tracht de titel „meesterpapegaai” te winnen.

(Mijn record is 19 - ik geef les aan papegaaien!)



```

5 REM PAPAPAPEGAARI
10 CLS : RESTORE : PLOT 100,90
20 FOR i=1 TO 7: READ x: READ
y: READ a: DRAW x,y,a: NEXT i
30 PLOT 89,139: DRAW 15,-10,-P
I
40 FOR i=0 TO 5 STEP .2: CIRCLE
E 95,131,i: NEXT i
100 DEF FN t()=INT ((65536*PEEK
23674+256*PEEK 23673+PEEK 23672
)/60): REM seconden sinds start
110 PRINT AT 14,0;"Ben je klaar
, papapapegaai. ?": IF INKEY$=""
THEN GO TO 110
120 LET a=97: LET b=25: LET t1=
FN t(): LET c=0
130 LET x$=CHR$ (a+b*RND): LET
a$=x$: PAUSE 1
140 PRINT AT 16,15;x$;
150 LET t2=FN t(): IF t2-t1>=10
THEN GO TO 200
160 LET x$=" ": PAUSE 10: IF IN
KEY$<>a$ THEN GO TO 140
170 LET c=c+1: GO TO 130
200 PRINT AT 16,5;"uw score : "
;c
210 LET t=INT (t2-t1): PRINT AT
19,5;"tijd : ";t;" seconden"
220 INPUT "Nog een keer? (j/n)?
";a$
230 IF a$="j" THEN GO TO 10
240 STOP
1000 DATA -5,8,.5,-8,-2,1,5,7,-1
,-10,-15,2,10,35,-3,40,5,-4,-15,
-40,.5

```


DUINDIGT

Dames en heren, de paarden zijn gestart! Red Arrow maakt een goede start ... Blondish boy geeft zijn jockey wat problemen ... en daar is de beroemde Spanish lady aan de binnenkant van de baan. Wat een spanning en dan zijn we nog maar net begonnen!

De eerste hindernis ... Kees Verkerk heeft nu de leiding overgenomen; op de hielen gevolgd door Speedy Conzales, ... Duindigt is een computer paardenrace. Bij dit spel kunnen niet zoveel paarden deelnemen als in werkelijkheid. We spelen hier steeds met drie paarden die ieder 10 afmattende ronden moeten lopen, U zult echter wel weten dat computer paarden werkelijk onvermoeibaar zijn.

Aan het begin van het spel worden de winkansen voor elk paard gegeven. U kunt nu inzetten. Zolang U geld heeft kunt U wedden.

Na het RUN commando zien we:

U KUNT INZETTEN
DE PAARDEN STAAN OP:
PAARD 1: 5 OP 1
PAARD 2: 4 OP 1
PAARD 3: 3 OP 1

De kansen kunnen van race tot race verschillen.

U HEEFT NOG ...GULDENS
UW INZET OP PAARD 1 IS ...

Als U geen geld genoeg heeft zegt de computer:

U LEEFT BOVEN UW STAND!

en als U er iets beter voorstaat:

U HEEFT NOG ... GULDEN

Direkt nadat U Uw inzet heeft gedaan verschijnt de renbaan op het scherm, met de paarden aan de start. De kansen en inzetten worden onderin het scherm getoond. Vervolgens ziet U:

START???

Toets ENTER in en daar gaan ze!

Als U niet op het verkeerde paard gewed heeft wordt het uitgekeerde bedrag aan Uw vermogen toegevoegd.

```

100 REM DUINDIGT
190 GO SUB 5000
200 REM hoofdprogramma
220 CLS
230 GO SUB 600
250 GO SUB 1300
265 BEEP .5,6
260 LET vm=(v(1)+v(2)+v(3))/3
270 FOR q=1 TO 3
280 GO SUB 500
290 NEXT q
300 BEEP .005,-3
310 IF finish<3 THEN GO TO 260
315 PRINT AT t(winner),10; FLAS
H 1;"DE WINNARR"
320 LET credit=credit+i(winner)
#p(winner)
325 FOR i=1 TO 500: NEXT i
330 IF credit>0 THEN GO TO 230
340 STOP
500 REM een beweging*****
505 LET p$="P"
510 IF d(q)=1 THEN RETURN
520 PRINT AT t(q),x(q);" "
530 IF s(q)=1 THEN LET s(q)=0:
GO TO 700
540 LET x(q)=x(q)+v(q)
550 LET v(q)=v(q)+vupg*RND*(vm-
v(q))
560 IF s(q)=2 THEN LET t(q)=t(q)
)+1: LET s(q)=0: GO TO 700
570 IF (xh1<x(q)) AND (x(q)<xh3
) THEN GO TO 670
580 IF x(q)<xmax THEN GO TO 700
590 LET x(q)=x(q)+2-xmax
600 LET r(q)=r(q)+1
610 PRINT AT t(q),0;r(q)
620 IF r(q)<rounds THEN GO TO 7
00

```

```

630 IF winner=0 THEN LET winner
=q
640 LET d(q)=1
650 LET finish=finish+1
660 GO TO 700
670 IF x(q)<xh2 THEN LET p$="Q"
: LET s(q)=1: GO TO 700
680 LET s(q)=2: LET t(q)=t(q)-1
: LET p$="R"
700 PRINT AT t(q),x(q); INK q;p
$
710 RETURN
800 REM init, paarden*****
810 LET p$="P"
820 LET winner=0: LET finish=0
830 FOR k=1 TO 3
840 LET x(k)=3
850 LET r(k)=0
855 LET p(k)=(INT (RND*5)+1)/5
860 LET v(k)=p(k)*dv+vmin
870 LET t(k)=5*(k-1)+3
880 LET d(k)=0
890 NEXT k
1000 REM kansen en inzetten*****
1010 LET ptot=0
1020 FOR i=1 TO 3
1030 LET p(i)=(v(i)-vmin)/dv
1040 LET ptot=ptot+p(i)
1050 NEXT i
1052 FOR k=1 TO 3
1054 LET p(k)=INT (ptot/p(k))
1056 NEXT k
1060 CLS
1070 PRINT AT 1,0;"*****
*****"
1080 PRINT "      U KUNT INZETTEN

1085 PRINT "*****
*****"
1090 PRINT
1100 PRINT "De paarden staan op:


1110 PRINT
1120 FOR i=1 TO 3
1130 PRINT "paard ";i;": ";p(i);
" op 1"
1135 NEXT i
1140 PRINT
1145 PRINT "U heeft nog: ";credi
t;" gulden
1150 FOR i=1 TO 3
1152 INPUT ("uw inzet op paard
";i;": is :");i(i)
1154 LET rest=credit-i(i)
1156 IF rest<0 THEN PRINT "U lee
ft boven uw stand!": GO TO 1152
1158 LET credit=credit-i(i): PRI
NT AT 11,0;"U heeft nog: ";credi
t;" gulden

```

```

1160 NEXT i
1200 RETURN
1300 REM zet baan op****
1310 CLS
1320 PRINT AT 16,0;"U heeft nog
";credit;" gulden"
1330 PRINT
1340 PRINT "          paard1  paa
rd2  paard3"
1350 PRINT "-----"
-----"
1360 PRINT "uitbetaling:";TAB 12
;p(1);TAB 19;p(2);TAB 26;p(3)
1370 PRINT "inzet ";TAB 12;i(1);
TAB 19;i(2);TAB 26;i(3)
1380 PRINT AT t(1)-3,0;"-----"
-----"
1390 FOR k=1 TO 3
1395 PRINT AT t(k),18;"x"
1400 PRINT AT t(k),0;r(k)
1410 PRINT AT t(k),x(k); INK k;"
p"
1420 PRINT AT t(k)+2,0;"-----"
-----"
1430 NEXT k
1440 INPUT "START???";a$
1450 BEEP .5,2
1460 RETURN
5000 REM initialisatie
5005 RANDOMIZE
5010 DIM r(3): DIM v(3): DIM x(3)
): DIM t(3): DIM d(3): DIM p(3):
DIM i(3): DIM s(3)
5020 LET vmin=2: LET dv=1: LET v
upg=2
5030 LET rounds=10: LET xmax=31
5040 LET credit=100
5050 LET xh=18: LET xh1=xh-3: LE
T xh2=xh-1: LET xh3=xh+1
5060 FOR i=0 TO 7: READ pa: POKE
USR "P"+i,pa: NEXT i

```



```

5070 FOR i=0 TO 7: READ pa: POKE
   USR "q"+i,pa: NEXT i
5080 FOR i=0 TO 7: READ pa: POKE
   USR "r"+i,pa: NEXT i
5090 RETURN
5100 DATA BIN 00000100,BIN 00000
111,BIN 00000111,BIN 10000100,BI
N 01111100,BIN 01111100,BIN 0100
0100,BIN 1000010
5110 DATA BIN 00000100,BIN 00000
111,BIN 00001100,BIN 00111100,BI
N 11111111,BIN 11100100,BIN 0100
0000,BIN 01100000
5120 DATA BIN 00000100,BIN 00000
111,BIN 00000110,BIN 00111100,BI
N 00111100,BIN 00100100,BIN 0100
0010,BIN 10000001

```


HET NIET ZO VERSCHRIKKELIJKE KLEURENPROBLEEM

Een tweedimensionale kubus? Onmogelijk zou U natuurlijk denken. Toch lijkt dit spel in veel opzichten op de zo befaamde kubus van Rubik.

Als het spel begint zien we 25 gekleurde vierkantjes die ogenschijnlijk willekeurig zijn verspreid op een bord van 5×5 . De rijen, respectievelijk kolommen zijn als volgt genummerd:

1					
2					
3					
4					
5					
	6	7	8	9	0


De horizontale rijen zijn dus genummerd van 1 tot 5 en de verticale rijen met 6, 7, 8, 9 en 0. Als we nu een rijnummer opgeven verschuiven alle vierkantjes in die rij één positie op. In een horizontale rij schuiven ze naar rechts als U een positief getal intypt en naar links als U een negatief getal intypt. Evenzo schuiven de vierkantjes in een kolom naar boven bij een positief getal en naar beneden bij een negatief getal. Als een vierkantje aan één kant als het ware van de rand af zou vallen verschijnt het weer aan de andere kant:



Het doel van dit alles? Wel we moeten de gekleurde vierkantjes in een zodanige positie trachten te brengen dat vijf horizontale rijen ieder met één bepaalde kleur ontstaan.

Wellicht zou het motto van dit spel kunnen zijn „sommige dingen lijken gemakkelijker dan ze zijn!”

```
5 REM HET NIET ZO VERSCHRIKKE
LIJKE KLEURENPROBLEEM
10 DIM a(6,6)
20 FOR x=1 TO 5
30 FOR y=2 TO 6
40 LET a(x,y)=y-1
50 NEXT y
60 NEXT x
70 FOR i=1 TO 20
80 LET k=1+INT (10*RND)
90 GO SUB 400
120 NEXT i
125 CLS
130 REM begin van het spel
140 FOR y=2 TO 6
150 LET p=(y-2)*4-1
160 PRINT AT (p+2),5;y-1
170 FOR i=1 TO 4
180 PRINT AT (p+i),7; INK a(1,y)
); "███"; INK a(2,y); "███"; INK
a(3,y); "███"; INK a(4,y); "███
"; INK a(5,y); "███"
190 NEXT i
200 NEXT y
```



```


210 PRINT AT 21,7;" 6    7    8
0    0"
220 INPUT AT 0,0;"Kolom nummer
(0..9): ";k;AT 1,0;"Rantal stepp
en (-4..4): ";s
225 IF k=0 THEN LET k=10
230 IF (k<1 OR k>10) OR (s<-4 OR
R s>4) THEN GO TO 220
240 IF s<0 THEN LET s=s+5
250 FOR i=1 TO s
270 GO SUB 400
280 NEXT i
290 GO TO 130
400 IF k>5 THEN GO TO 470
420 FOR x=6 TO 2 STEP -1
430 LET a(x,k+1)=a(x-1,k+1)
440 NEXT x
450 LET a(1,k+1)=a(6,k+1)
460 RETURN
470 FOR y=2 TO 6
480 LET a(k-5,y-1)=a(k-5,y)
490 NEXT y
500 LET a(k-5,6)=a(k-5,1)
510 RETURN

```


QUI VIVE

Op je qui vive zijn wil zeggen waakzaam zijn. Dit spel wordt Qui Vive genoemd omdat je altijd alert moet zijn op winnende situaties. De regels zijn in feite opvallend eenvoudig, maar het programma dat nodig is om deze regels te „verwoorden” is echter allesbehalve eenvoudig. Met recht is het voor ieder spelletjesprogrammeur een uitdaging om een dergelijk programma op te stellen, of uitgaande van dit programma verbeterde strategieën te ontwikkelen. Qui Vive is ontworpen door Eugene de Wolf en het spel wordt gespeeld op een 5×5 bord. Elke speler heeft 5 stenen en moet trachten ze zodanig te plaatsen dat één van de 7 symmetrische patronen ontstaat en wel:


een horizontale lijn




een verticale lijn




een diagonale lijn




een recht kruis




een scheef kruis



een horizontale vleugel



een verticale vleugel



Aan het begin van het spel is het bord leeg en worden om de beurt de stenen geplaatst. De velden op het bord worden als volgt aangeduid:

⑤	1	2	3	4	5
④	6	7	8	9	10
③	11	12	13	14	15
②	16	17	18	19	20
①	21	22	23	24	25
	Ⓐ	Ⓑ	Ⓒ	Ⓓ	Ⓔ

Om een steen op een bepaald veld te plaatsen hoeven we slechts de coördinaten van dat veld in te voeren; bijvoorbeeld voor veld 21 toetsen we A1 in, voor veld 12 B3, etc. Uw stenen worden aangeduid met P en die van de computer met C.

Als alle 10 stenen eenmaal op het bord geplaatst zijn en niemand heeft nog gewonnen, gaat het spel verder door ombeurten een steen te verplaatsen en zo te trachten een winnende positie te bereiken.

Om een steen te verplaatsen moet U eerst de coördinaten van het veld waarop die steen staat invoeren en vervolgens die van het veld waarheen U de steen wilt verplaatsen. Uiteraard moet U voortdurend attent zijn op mogelijke winnende patronen en tegelijkertijd de mogelijk winnende patronen van de computer blokkeren.

Dit programma is stellig een van de meest interessante programma's in dit boek. Het bevat o.a. alle mogelijke (42) winnende patronen en meer dan 100 bepaalde strategisch sterke opstellingen waarmee een of twee winnende patronen gemaakt kunnen worden.

Zeer veel strategische problemen die het opstellen van schaakprogramma's zo uitdagend maken treft men ook in dit programma aan.

Zoals altijd met dit soort programma's; er zijn nog vele varianten mogelijk. Zo dagen we iedere programmeur uit om het spel aanzienlijk te versnellen, want dit blijft doorgaans toch het belangrijkste probleem bij het programmeren in BASIC van dergelijke interessante spelen.



```
10 REM **QUI VIVE**
20 DIM s(42,5): DIM o(42)
30 DIM d(108,2): DIM f(16,9)
40 DIM e(5,2): DIM b(25): DIM
c(25)
50 GO TO 3000
100 REM **PUNTEN BEREKENEN VAN
DUBBELE FIGUREN (f)**
110 LET p1=0: LET p2=0: LET c1=
0: LET c2=0
120 LET d1=b(f(n,1)): LET d2=b(
f(n,2))
130 IF d1=1 OR d2=1 THEN LET p1
=1
140 IF d1=10 OR d2=10 THEN LET
c1=10
150 LET d1=b(f(n,3)): LET d2=b(
f(n,4))
160 IF d1=1 OR d2=1 THEN LET p2
=1
170 IF d1=10 OR d2=10 THEN LET
c2=10
180 LET gv=b(f(n,5))+b(f(n,6))+
b(f(n,7))
```



```

190 LET p=gv+p1+p2+c1+c2
200 IF gv=30 AND (c1=0 OR c2=0)
THEN LET p=p-10
210 IF gv=3 AND (p1=0 OR p2=0)
THEN LET p=p-1
220 RETURN
800 REM **BEREKEN VELDEN VAN DU
BBELE FIGUREN**
810 LET tc=5: LET tn=1
820 FOR i=1 TO 5: LET c(s(d(f,1),i))=1: NEXT i
830 FOR i=1 TO 5: LET s=s(d(f,2),i)
840 IF c(s)=1 THEN GO TO 860
850 LET f(n,tn)=s: LET tn=tn+1:
GO TO 870
860 LET f(n,tc)=s: LET tc=tc+1:
LET c(s)=0
870 NEXT i
880 FOR i=1 TO 5: LET s=s(d(f,1),i)
890 IF c(s)=1 THEN LET f(n,tn)=s:
LET tn=tn+1: LET c(s)=0
900 NEXT i: RETURN
1100 REM **BEPAAAL EIGENAAR VAN E
NKELE FIGUREN (s)**
1110 LET om=-1: LET wm=-1
1120 FOR s=1 TO 42
1130 LET p=b(s(s,1))+b(s(s,2))+b
(s(s,3))+b(s(s,4))+b(s(s,5))
1140 IF p=0 OR p=10 OR p=20 OR p
=30 THEN LET o(s)=10: GO TO 1240
1150 IF p<>40 THEN GO TO 1190
1160 GO SUB 1250
1170 GO SUB 1300
1180 LET wm=1: LET s=42: GO TO 1
240
1190 IF p>20 OR p<3 OR p=11 OR p
=12 THEN LET o(s)=0: GO TO 1240
1200 LET o(s)=1
1210 IF p=4 AND om=-1 THEN GO SU
B 1250
1220 IF p=14 THEN GO SUB 1300
1230 IF p=5 THEN LET s=42
1240 NEXT s: RETURN
1250 REM **BEPAAAL VERPLICHTE ZET
(om)**
1260 FOR i=1 TO 5
1270 IF b(s(s,i))=0 THEN LET om=
s(s,i)
1280 NEXT i: RETURN
1300 REM **BEPAAAL GEPEENDE STEEN*
1310 FOR i=1 TO 5
1320 IF b(s(s,i))<>10 THEN GO TO
1360
1330 FOR j=1 TO cc
1340 IF s(s,i)=e(j,1) THEN LET e
(j,2)=1
1350 NEXT j

```



```


1360 NEXT i
1370 RETURN
1400 REM **BEPAAL RELEVANTE DUBB
ELE FIGUREN**
1410 LET nc=0: LET np=0: LET n=1
1420 FOR i=1 TO 16
1430 LET f(i,9)=-1
1440 LET f(i,8)=0
1450 NEXT i
1460 FOR f=1 TO 108
1470 LET o1=o(d(f,1)): LET o2=o(
d(f,2))
1480 IF o1<>o2 OR o1=0 OR o2=0 T
HEN GO TO 1600
1490 GO SUB 300
1500 GO SUB 100
1510 IF p=4 OR p=14 THEN LET f(n
,8)=1: GO TO 1540
1520 IF o1=1 OR p<=f(n,9) THEN G
O TO 1600
1530 LET f(n,8)=10
1540 LET f(n,9)=p
1550 LET n=1
1560 FOR i=1 TO 16
1570 IF f(i,9)>=f(n,9) THEN GO T
O 1590
1580 IF f(i,8)=0 OR f(i,8)=10 TH
EN LET n=i
1590 NEXT i
1600 NEXT f
1610 RETURN
1700 REM **BEPAAL STRATEGISCHE
WAARDE ZET**
1710 IF wm=1 THEN GO TO 1860
1720 LET mc=0: LET n2=0
1730 FOR n=1 TO 16
1740 IF f(n,8)=0 THEN GO TO 1800
1750 GO SUB 100
1760 IF f(n,8)=1 THEN GO TO 1790
1770 IF p=mc THEN LET n2=n2+1
1780 IF p>mc THEN LET mc=p: LET
n2=1
1790 IF p=4 THEN LET n=16
1800 NEXT n

```

```

1810 IF r=0 THEN GO TO 1860
1820 IF p=4 THEN RETURN
1830 IF mc<mp THEN RETURN
1840 IF mc=mp AND n2<n1 THEN RETURN
1850 IF mc=mp AND n2=n1 AND AND<
.5 THEN RETURN
1860 LET mp=mc: LET n1=n2
1870 LET r=d: LET t=b
1880 RETURN
1900 REM **BEURT VAN DE SPELER**
1910 PRINT AT 17,6;"NU IS HET UW
BEURT! "; PRINT AT 19,0;TAB
31;" "; PRINT AT 20,0;TAB 31;" ";
1920 IF pc<5 THEN GO TO 1960
1930 INPUT "WELKE STEEN PAKT U O
P?";TAB 32;"A1..E5:- ";x$
1940 GO SUB 2030: LET t=x: IF np
OR b(t)<>1 THEN BEEP 1,20: GO T
O 1930
1950 PRINT AT 19,3;"U NEEMT EEN
STEEN VAN: ";x$
1960 INPUT "WAAR ZET U DE STEEN
NEER? A1..E5: ";x$
1970 GO SUB 2030: LET r=x: IF np
OR b(r)<>0 THEN BEEP 1,20: GO T
O 1960
1980 PRINT AT 20,3;"U ZET DE STE
EN OP : ";x$
1990 IF pc=5 THEN LET b(t)=0
2000 IF pc<5 THEN LET pc=pc+1
2010 LET b(r)=1: GO SUB 4000
2020 RETURN
2030 REM **INVOER**
2040 LET np=0: LET x=1: IF LEN x
<>2 THEN GO TO 2090
2050 LET co=CODE x$(1): IF co>96
THEN LET x$(1)=CHR$ (co-32)
2060 LET l$=x$(1): LET d$=x$(2)
2070 IF l$<"A" OR l$>"E" OR d$<"
1" OR d$>"5" THEN GO TO 2090
2080 LET x=CODE l$-5*VAL d$-39:
RETURN
2090 LET np=1: RETURN
2200 REM **BEURT VAN DE COMPUTER
2210 BEEP 1,30: IF cc<5 THEN LET
t=cc+1: LET cc=cc+1: GO TO 2270
2220 PRINT AT 19,4;"IK NEEM EEN
STEEN VAN: ";CHR$ (64+t-INT ((t-
1)/5)*5);5-INT ((t-1)/5)
2230 LET b(t)=0
2240 FOR c=1 TO 5
2250 IF e(c,1)=t THEN LET t=c: L
ET c=5
2260 NEXT c
2270 PRINT AT 20,4;"IK ZET DE ST
EEN OP : ";CHR$ (64+r-INT ((r-1)
/5)*5);5-INT ((r-1)/5)
2280 LET b(r)=10: LET e(t,1)=r

```



```


2290 GO SUB 4000
2300 RETURN
3000 REM **HOOFDPROGRAMMA**
3005 LET d=1: LET b=1: LET wm=-1
3010 GO SUB 4100: PRINT AT 17,6;
"EVEN GEDULD A.U.B."
3020 GO SUB 5000: GO SUB 6000
3030 LET cc=0: LET pc=0: PRINT A
T 17,0;TAB 31;""
3040 INPUT "WILT U BEGINNEN?";TA
B 32;"J OF N:";a$
3050 IF a$="N" OR a$="n" THEN LE
T r=13: GO TO 3270
3060 IF a$="J" OR a$="j" THEN GO
TO 3290
3065 GO TO 3040
3070 FOR i=1 TO 5: LET e(i,2)=0:
NEXT i
3080 GO SUB 1100: IF p=5 THEN GO
TO 3400
3090 PRINT AT 17,4;" NU IS HET M
IJN BEURT! ": PRINT AT 19,0;TAB
31;"": PRINT AT 20,0;TAB 6;"EVEN
GEDULD A.U.B.";TAB 31;""
3100 IF wm<>1 THEN GO SUB 1400
3110 LET mp=0: LET n1=0: LET r=0
3120 LET bf=1: LET ef=25
3130 IF om<>-1 THEN LET bf=om: L
ET ef=om
3140 FOR d=bf TO ef
3150 IF b(d)<>0 THEN GO TO 3260
3170 LET b(d)=10
3180 IF cc<5 THEN GO SUB 1700: G
O TO 3250
3190 FOR c=1 TO 5
3200 IF e(c,2) THEN GO TO 3240
3210 LET b=e(c,1): LET b(b)=0

```

```

3220 GO SUB 1700
3230 LET b(b)=10
3240 NEXT c
3250 LET b(d)=0
3260 NEXT d
3270 GO SUB 2200
3280 IF wm=1 THEN GO TO 3410
3290 GO SUB 1900
3300 GO TO 3070
3400 PRINT AT 21,7; PAPER 6;"U H
EEFT GEWONNEN!"; GO TO 3420
3410 PRINT AT 21,8; PAPER 5;"IK
HEB GEWONNEN!"
3420 BEEP 2,25: STOP
4000 REM **STENEN AFBEELDEN**
4010 FOR i=0 TO 4: FOR j=1 TO 5
4020 LET h=5*i+j
4030 IF b(h)=0 THEN PRINT AT 4+2
*i,9+2*j;" ";
4040 IF b(h)=1 THEN PRINT AT 4+2
*i,9+2*j; PAPER 6;"P";
4050 IF b(h)=10 THEN PRINT AT 4+
2*i,9+2*j; PAPER 5;"C";
4060 NEXT j: NEXT i: RETURN
4070 PRINT AT 14,11;"A B C D E";
4080 RETURN
4100 REM **BORD TEKENEN**
4110 PRINT AT 1,8;"*** QUIVIVE *
**"
4120 FOR i=0 TO 5
4130 PLOT 84,147-16*i: DRAW 80,0
4140 PLOT 84+16*i,147: DRAW 0,-8
0
4150 NEXT i
4160 FOR i=1 TO 5: PRINT AT 14-2
*i,21;i;: NEXT i
4170 PRINT AT 14,11;"A B C D E";
4180 RETURN
5000 REM **INITIALISATIE ENKELVO
UDIGE FIGUREN**
5010 FOR i=1 TO 42: READ x$
5020 FOR j=1 TO 5
5030 LET s(i,j)=CODE x$(j)-64
5040 NEXT j: NEXT i: RETURN
5050 REM VLEUGELS (AV><)
5060 DATA "UQMSY","PLHNT","KGCIO
"
5070 DATA "AGMIE","FLRNJ","KQUSO
"
5080 DATA "AGMOU","BHNRV","CIOSU
"
5090 DATA "EIMSY","DHLRX","CGKQW
"
5100 REM MAAL (X)
5110 DATA "ACGKM","BDHLN","CEIMO
"
5120 DATA "FLHPR","GIMQS","HJNRT
"
5130 DATA "KMOUW","LNRUX","MOSWY
"

```



```

5140 REM PLUS (+)
5150 DATA "BFGHL", "CGHIM", "DHIJN
"
5160 DATA "GKLMQ", "HLMNR", "IMNOS
"
5170 DATA "LPQRV", "MQRSW", "NRSTX
"
5180 REM DIAGONAAL (/ \)
5190 DATA "AGMSY", "EIMQU"
5200 REM VERTIKAAL (|)
5210 DATA "AFKPU", "BGOLU", "CHMRW
"
5220 DATA "DINSX", "EJOTY"
5230 REM HORIZONTAL (-)
5240 DATA "ABCDE", "FGHIJ", "KLMNO
"
5250 DATA "PQRST", "UVVXY"
6000 REM **INITIALISATIE DUBBELE
FIGUREN**
6010 FOR i=0 TO 107 STEP 12: REA
D X$
6020 FOR j=1 TO 12
6030 LET d(i+j,1)=CODE X$(2*j-1)
-48: LET d(i+j,2)=CODE X$(2*j)-4
8
6040 NEXT j: NEXT i
6050 RETURN
6060 DATA "171:1A1C1E1M1O1P2>2@2
B2J"
6070 DATA "393<3=3?3G474:4=4?4A4
G40"
6080 DATA "4PS@5B5D5J696<6C6E6M7
=7A"
6090 DATA "7C7I7O7P8>8B8D8J9?9E9
K:?"
6100 DATA " :A:E:K:O:P;>;@;D;J<=<
C<I"
6110 DATA " =G=I=O>F>H>J?G?K?P@F@
J@L"
6120 DATA "AGAIKAMAQAPBHBUBNCIC
MCP"
6130 DATA "DJDLDNEKEMEOPRFUGSGWH
THU"
6140 DATA "IRIXJSJXKTKXLRLYMSMYN
TNY"

```


S.T.M.

Een van de interessantste onderwerpen uit de psychologie is ongetwijfeld ons geheugen. De functie van het geheugen wordt wel eens omschreven als „te reageren op actuele situaties op grond van ervaringen”. We laten deze definitie maar snel voor wat hij is. Waarom vergeten we sommige dingen en waarom herinneren we ons bepaalde andere zaken? We kunnen iemand die we jaren niet gezien hebben onmiddellijk herkennen, tevens kunnen we een melodie herkennen waarvan we slechts een paar noten horen, terwijl we aan de andere kant een telefoonnummer vergeten.

In de wetenschap worden gewoonlijk twee typen geheugen onderscheiden n.l. het korte termijn geheugen (Short Term Memory of S.T.M.) en het lange termijn geheugen (Long Term Memory of L.T.M.). Wilt U hierover meer weten dan kunnen we U het boek aanraden: Introduction to psychology van Hilgard en Atkinson.

Dit typische competitie spel test ons korte termijn geheugen (S.T.M.) en wel door zeer kort verschillende letters te tonen die we dan moeten onthouden. U zult stellig merken hoe moeilijk het wel is veel letters te onthouden. Ga nu met Uw vrienden of familie na wie de grootste opslagbak onder zijn pet heeft....

Op het scherm krijgen we eerst het volgende „menu” te zien:

WILT U SPELEN MET:


1. LETTERS?
2. CIJFERS?
3. ALLEEN MET 0 EN 1
4. OF WILT U OPHOUDEN?

We kunnen nu de keus maken door 1, 2, 3 of 4 in te typen. Stel dat we nu 1 intypen, we kiezen er dan kennelijk voor met letters te spelen. Een bepaalde letter zal slechts even op het scherm zichtbaar zijn. Deze letter moet U intypen. De computer zal vervolgens 2 letters tonen die we ook moeten intypen, vervolgens 3, enz. Het is duidelijk dat hoe groter het aantal letters wordt hoe moeilijker het zal zijn om alles te onthouden. Hoeveel kunt U onthouden? Kunt U ons record van acht letters verbeteren?!

```

5 REM STM
10 CLS : PRINT "WILT U SPELEN
MET: "
20 PRINT " 1. LETTERS?"
30 PRINT " 2. CIJFERS?"
40 PRINT " 3. ALLEEN MET 0 EN
1?"
50 PRINT " 4. OF WILT U OPHOUD
EN?"
55 PRINT : PRINT
60 LET a$=INKEY$: IF a$<"1" OR
a$>"4" THEN GO TO 60
70 PAUSE 50: LET x=VAL a$: GO
TO 100*x
100 LET a=26: LET b=97: GO TO 3
20

```



```

200 LET a=10: LET b=48: GO TO 3
20
300 LET a=2: LET b=48: PAUSE 1
320 DEF FN t()=INT ((65536*PEEK
23874+256*PEEK 23873+PEEK 23872
)/60): REM seconden sinds start
330 LET c=1: LET t1=FN t()
340 PAUSE 10: LET a$="": FOR i=
1 TO c: LET b$=CHR$ INT (b+RND*a
): PRINT AT 10,i;" "+b$): LET a$
=a$+b$
350 PAUSE 15: NEXT i: PRINT AT
10,i;" "

```

```

360 INPUT "ANTWOORD: ";b$: IF b
$=a$ THEN GO TO 390
370 PRINT : PRINT "FOUT, UW SCO
RE IS: ";c-1: PRINT "HET WAS ";a
$;" , NIET ";b$
380 LET t2=FN t(): PRINT "TIJD:
";INT (t2-t1);" SECONDEN": INPU
T "NOG EEN KEER (J/N)?";a$: IF a
$="J" THEN GO TO 10
385 GO TO 400
390 LET c=c+1: LET t2=FN t(): I
F t2-t1>=40 THEN PRINT : PRINT "
UW SCORE IS: ";c-1: GO TO 380
395 GO TO 340
400 STOP

```


VAN 5 NAAR 0

Dit uitdagende spel zagen we voor het eerst op een Casio rekenmachine. Het is werkelijk een boeiend spel dat zeer veel logisch denkwerk vereist.

Het spel wordt gespeeld op een bord met 9 vakken. Deze zijn als volgt genummerd:

1	2	3
4	5	6
7	8	9

Door ieder cijfer kunnen we nu een horizontale en een verticale lijn trekken en deze lijnen zullen zo in totaal nog 4 andere vakjes kruisen. Bijvoorbeeld de lijnen getrokken door positie 4 zullen ook de posities 1, 5, 6 en 7 kruisen.



De posities op het speelbord worden ogenschijnlijk bezet door een willekeurig patroon van cijfers tussen 0 en 5. Als U nu een van de posities

aangeeft (op het scherm wordt getoond hoe dit moet), zullen alle cijfers op zowel de horizontale als de verticale lijn verhoogd worden met 1 (behalve 5, dat wordt 0). Bijvoorbeeld, als we uitgaan van

2	2	4
1	5	2
1	5	4

en we geven vakje 4 aan, dan zal het bord veranderen in

3	2	4
2	0	3
2	5	4

De puzzel is opgelost als we uiteindelijk de volgende situatie bereiken:

0	0	0
0	0	0
0	0	0


```

10 REM VAN 5 NAAR 0
100 DIM a(3,3)
105 LET c$="123456789"
108 PRINT AT 8,6;"KIEZEN";TAB 1
6;"BORD"
110 PRINT AT 10,6;"-----";TAB 1
6;"-----"
120 FOR i=1 TO 3
125 LET i2=i*2: LET i3=(i-1)*3
130 PRINT AT i2+9,5;"1";i3+1;"
";i3+2;" ";i3+3;"1";TAB 15;"1"
135 PRINT AT i2+10,5;"1" "1";
TAB 15;"1" "1"
140 NEXT i
150 PRINT AT -15,5;" ----- ";TAB
15;" -----"
160 FOR a=1 TO 10
170 LET i=INT (RND*9)+1: GO SUB
400
180 NEXT a
190 LET m=0: LET t=0
200 LET t$=INKEY$
210 IF t$="" THEN GO TO 200
220 FOR i=1 TO 9
230 IF t$=c$(i) THEN GO TO 260
240 NEXT i
250 GO TO 200
260 LET m=m+1
270 GO SUB 400
280 IF t<9 THEN GO TO 200
290 PRINT FLASH 1;"GELUKT !"
300 PRINT : PRINT "IN ";m;" ZET
TEN"
310 STOP
400 LET x=INT ((i-1)/3)+1
410 LET y=i-3*(x-1)
420 LET a(x,y)=a(x,y)-1
430 FOR i=1 TO 3
440 LET a(x,i)=a(x,i)+1
450 LET a(i,y)=a(i,y)+1
460 NEXT i
470 LET t=0
480 FOR i=1 TO 3
490 FOR j=1 TO 3
500 IF a(i,j)=0 THEN LET a(i,j)
=0
510 IF a(i,j)=0 THEN LET t=t+1
520 PRINT AT 11+(i-1)*2,16+(j-1)
)*2;a(i,j)
530 NEXT j
540 NEXT i
550 RETURN

```


ESCHER

Het zo boeiende werk van de Nederlandse graficus M.C. Escher (1902 - 1972) is gebaseerd op symmetrie en wiskundige schema's en het is wellicht dit feit dat zoveel computerprogrammeurs aanspreekt. Met behulp van dit programma kunt U Uw eigen „Escher" maken en het aardige is dat het is gebaseerd op dezelfde principes die Escher gebruikt. Laten we als voorbeeld eens het volgende patroon bekijken:




Deze voorstelling is gebaseerd op een mozaiek in het Alhambra in Spanje (Escher bestudeerde dergelijke patronen zeer zorgvuldig). Als we goed kijken kunnen we zien dat alle delen even groot zijn en zo zijn geordend dat de zwarte figuren hetzelfde patroon, gedraaid over een rechte hoek, opleveren als de witte figuren.


Escher bestudeerde dergelijke patronen en ontwikkelde zelfs een spel gebaseerd op dit principe. Hij ontwierp n.l. een stempel zoals:



Een dergelijk stempel kunnen we op vier manieren afbeelden



We kunnen de patronen bijvoorbeeld zodanig rangschikken dat ze een attractief doorlopend patroon vormen.



Met dit programma kunnen we in feite iets soortgelijks doen. Nadat U een basisvierkant heeft gevuld met symbolen zal de computer dit draaien en verschuiven om zo het scherm met Uw eigen „Escher” te vullen. Aan het begin verschijnt op het scherm:

U MOET HET BASISVIERKANT VULLEN VAN 16 BIJ 16.
GEBRUIK ALLEEN 1(INKT) EN 0(LEEG)

De computer voert U dan rij voor rij door het vierkant en elke rij moet gevuld worden met een patroon van 16 énen en nullen in elk gewenst patroon. Als alle rijen gevuld zijn, is de computer even bezig. Na de pieptoon echter kunt U Uw basisvierkant draaien met behulp van de cursorbesturingstoetsen.

Probeer het maar! U zult verbaasd staan over de prachtige patronen die U maken kunt.

```

10 REM ESCHER
100 DIM a$(16,16): DIM b$(16,16)
)
110 DEF FN c(l$)=128*VAL l$(1)+
64*VAL l$(2)+32*VAL l$(3)+16*VAL
l$(4)+8*VAL l$(5)+4*VAL l$(6)+2
*VAL l$(7)+VAL l$(8)
120 LET cb=144
200 REM figuur invullen
210 PRINT "U moet een basisvier
kant van 16 bij 16 vullen"
230 PRINT "Gebruik alleen 1_(in
kt) en 0 (leeg)!"
240 PRINT
250 FOR i=1 TO 16
260 INPUT ("rij ";i);a$(i)
270 PRINT TAB 8;a$(i)
280 NEXT i
300 REM karakters definieren
310 FOR b=144 TO 159 STEP 4
320 FOR i=0 TO 7
330 POKE USR CHR$(b)+i,FN c(a$(
i+1, TO 8))
340 POKE USR CHR$(b+1)+i,FN c(
a$(i+1,9 TO ))
350 POKE USR CHR$(b+2)+i,FN c(
a$(i+9, TO 8))
360 POKE USR CHR$(b+3)+i,FN c(
a$(i+9,9 TO ))
370 NEXT i
380 REM karakters roteren
390 FOR i=1 TO 16
400 FOR j=1 TO 16

```

```

410 LET b$(i,j)=a$(j,17-i)
420 NEXT j
430 NEXT i
440 REM naar oorspronkelijke ar
ray kopiëren
450 FOR i=1 TO 16
460 LET a$(i)=b$(i)
470 NEXT i
480 NEXT b
500 REM karakters printen
510 PRINT "U kunt beginnen"
520 BEEP .5,2: CLS
530 LET ro=10: LET co=15
540 PRINT AT ro,co;CHR$ 144;CHR
$ 145
550 PRINT AT ro+1,co;CHR$ 146;C
HR$ 147
560 LET t$=INKEY$
570 LET rj=10*(t$="5")+20*(t$="
6")+30*(t$="7")+40*(t$="8")
580 IF rj=0 THEN GO TO 560
590 GO SUB 690+rj
600 PRINT AT ro,co;CHR$ ch;CHR$
(ch+1)
610 PRINT AT ro+1,co;CHR$ (ch+2
);CHR$ (ch+3)
620 GO TO 560
700 LET ch=148: LET co=co-2*(co
>2): RETURN
710 LET ch=152: LET ro=ro+2*(ro
<19): RETURN
720 LET ch=144: LET ro=ro-2*(ro
>2): RETURN
730 LET ch=156: LET co=co+2*(co
<29): RETURN

```

HET REKENWONDER

Hoe goed bent U in rekenen? Bij dit spel toont de computer ons steeds 6 getallen en een groter „doelgetal”. U mag twee getallen uitkiezen plus één van de vier basisbewerkingen van het rekenen, dus:

optelling +
aftrekken —
vermenigvuldigen *
delen /

Met de getallen die U dan gekozen heeft, wordt dan de gekozen bewerking uitgevoerd. Als U bijvoorbeeld $7 + 18$ heeft ingevoerd, berekent de computer $7 + 18 = 25$. De getallen 7 en 18 worden dan in de oorspronkelijke reeks vervangen door het getal 25. Nu kiest U opnieuw, enz. Het doel van dit alles is om te eindigen met een getal in Uw lijst dat gelijk is aan het doelgetal of hier zo dicht mogelijk bijkomt.

Rekenwonder is geen ingewikkeld spel. Het is zeker geschikt voor het gehele gezin en misschien is er wel een rekenwonder in Uw midden ...

```
10 REM HET REKENWONDER
20 RANDOMIZE
30 LET t=0
40 LET x=0
50 LET r=0
60 DIM d(7)
70 GO TO 1000
100 REM initialisatie
110 CLS
120 FOR i=1 TO 7
130 LET d(i)=INT (RND*10+1)
140 IF RND<.25 THEN LET d(i)=25
150 NEXT i
160 LET f=INT (RND*900+99)
170 RETURN
200 REM scherm layout
210 CLS : PRINT "***** HET REK
ENWONDER *****"
220 PRINT AT 2,0;"PROBEER DIT G
ETAL TE MAKEN: ";f
```

```

230 PRINT AT 4,0;"GEBRUIK DEZE·
GETALLEN :: "
240 FOR i=1 TO 7
250 IF d(i)<>0 THEN PRINT d(i);
";
260 NEXT i
280 RETURN
400 REM bewerking invoeren
410 INPUT "BEWERKING? ";f$
420 PRINT AT 8,5;f$;
430 FOR n=2 TO LEN f$
440 IF f$(n)<"0" THEN GO TO 470
450 NEXT n
460 GO TO 400
470 LET a=VAL f$( TO n-1)
480 LET b=VAL f$(n+1 TO )
490 IF VAL f$<>INT VAL f$ THEN
GO TO 400
500 RETURN
600 REM verkeerde getallen
610 PRINT AT 10,5;"U GERUIKTE U
ERKEERDE GETALLEN !"
620 PRINT AT 11,5;"GEEN PUNTEN
IN DEZE RONDE"
630 RETURN
700 REM juiste getallen
710 PRINT AT 10,5;"U HEEFT: ";U
AL f$;" OP ";f;
720 LET x=5-ABS (f-VAL f$)
730 IF x<0 THEN LET x=0
740 IF x=5 THEN PRINT AT 11,9;
FLASH 1;"GOED GEDAAN"
750 LET t=t+x
760 PRINT AT 12,5;"U KRIJGT ER
";x;" PUNTEN BIJ"
770 RETURN
900 REM einde
910 PRINT AT 19,0;"          NU HEEF
T U ";t;" PUNTEN "
920 PRINT AT 20,0;"          IN ";f;
" RONDEN
930 INPUT "NOG EEN KEER? (J of
n) ";z$
940 IF z$="J" THEN GO TO 1000
950 STOP
1000 REM hoofdprogramma
1005 LET r=r+1
1010 GO SUB 100
1020 GO SUB 200
1030 GO SUB 400
1040 REM controle
1050 FOR i=1 TO 7
1060 IF d(i)=a THEN GO TO 1090
1070 NEXT i
1080 GO TO 1200
1090 LET d(i)=0
1100 FOR i=1 TO 7
1110 IF d(i)=b THEN GO TO 1140

```



```
1120 NEXT i
1130 GO TO 1200
1140 LET d(i)=VAL f$
1145 PRINT "=";VAL f$
1150 PRINT
1160 INPUT "KLARR? (j of n) ";z$
1170 IF z$="j" THEN GO TO 1300
1180 GO TO 1020
1200 GO SUB 600
1210 GO SUB 900
1220 GO TO 1000
1300 GO SUB 700
1310 GO TO 1210
```

DE SHAKESPEARE SHUFFLE

Shakespeare, 's werelds beroemste schrijver, en schaken, de koningin onder de spelen, worden in deze ongewone puzzel gecombineerd. Op een schaakbord zijn de volgende letters en cijfers geplaatst:


T	O		B	E		O	R
N	O	T		T	O		B
E		T	H	A	T		I
S		T	H	E		Q	U
E	S	T	I	O	N	.	
W	I	L	L	I	A	M	
S	H	A	K	E	S	P	E
A	R	E		I	6	O	3

Er is slechts één schaakstuk n.l. het paard en dat kan zich over het bord bewegen als in ieder schaakspel. Welnu als dit paard een aantal sprongen uitvoert zal op het bord de tekst door elkaar gehusseld worden. Aan het begin van het spel zien we:

AANTAL SPRONGEN (1 - 100)?

Nu geven we het aantal sprongen aan die het paard van te voren moet maken. Het zal duidelijk zijn dat hoe hoger dit getal is, hoe verwarrender is gehusseld. Uw taak is nu de oorspronkelijke tekst te herstellen door het paard weer bepaalde sprongen te laten uitvoeren.

Om een bepaalde sprong aan te geven typen we steeds een getal in en wel volgens onderstaande figuur.

	1		2	
8				3
				
7				4
	6		5	

```


5 REM DE SHAKESPEARE SHUFFLE
10 DIM x(8): DIM y(8)
20 RANDOMIZE
30 GO TO 4000
40 REM initialisatie
50 LET x0=4: LET y0=4
60 LET b$="TO BE ORNOT TO BE T
HAT IS THE QUESTION. WILLIAM SHA
KESPEARE 1603"
70 LET a$=b$
80 LET m$=" 1 2 3   3 K 7
4 6 5 "
90 FOR k=1 TO 8
100 READ x(k),y(k)
110 NEXT k
120 RETURN
500 REM zet
510 IF begin<>1 THEN RETURN
520 LET m=INT (RND*8+1)
530 LET xs=x(m): LET ys=y(m)
540 IF x0+xs>8 OR x0+xs<1 OR y0
+ys>8 OR y0+ys<1 THEN GO TO 510
550 REM letters verwisselen
560 LET p=x0+xs+(y0+ys-1)*8
570 LET v=x0+(y0-1)*8
580 LET w$b$(p)
590 LET b$(p)=b$(v)
600 LET b$(v)=w$
610 LET x0=x0+xs: LET y0=y0+ys
620 RETURN
1000 REM tekst laten zien
1010 FOR k=1 TO 8: FOR j=1 TO 8
1020 PRINT AT 1+2*k,14+2*j;b$((k
-1)*8+j);
1030 NEXT j: NEXT k
1040 RETURN
1500 REM positie laten knipperen
1510 PRINT AT 1+2*y0,14+2*x0; FL
ASH 1; OVER 1;" ";
1520 RETURN

```

```

2000 REM bord tekenen
2010 FOR i=1 TO 17 STEP 2
2020 PLOT 8*i+116,155
2030 DRAW 0,-128
2040 PLOT 124,163-8*i
2050 DRAW 128,0
2060 NEXT i
2070 PRINT AT 0,21;"BORD";
2080 RETURN
2500 REM mogelijke zetten tonen
2510 FOR i=1 TO 11 STEP 2
2520 PLOT 8*i-4,139
2530 DRAW 0,-80
2540 PLOT 4,147-8*i
2545 DRAW 80,0
2550 NEXT i
2560 FOR k=1 TO 5: FOR j=1 TO 5
2570 PRINT AT 3+2*k,2*j-1;m$(k-
1)*5+j);
2580 NEXT j: NEXT k
2590 PRINT AT 1,3;"ZETTEN";
2600 RETURN
4000 REM hoofdprogramma
4010 GO SUB 40
4020 INPUT "AANTAL SPRONGEN (1-1
00): ";l: IF l<1 OR l>100 THEN G
O TO 4020
4030 LET begin=1
4040 FOR k=1 TO l: GO SUB 500: N
EXT k
4050 GO SUB 2000: GO SUB 1000: G
O SUB 2500
4060 PRINT AT 20,3;"AANTAL SPRON
GEN=";l;
4070 LET move=0: LET begin=0

```



```

4080 GO SUB 1500
4090 INPUT "      ZET: ";m
4100 IF m<1 OR m>8 THEN GO TO 40
90
4110 LET move=move+1: GO SUB 530
4120 GO SUB 1000: IF b#<>a# THEN
  GO TO 4080
4127 IF move=1 THEN LET z$=" ZET
"
4128 IF move<>1 THEN LET z$=" ZE
TTEN"
4130 PRINT AT 21,0;"GELUKT! IN "
;move;z$;
4140 DATA -1,-2,1,-2,2,-1,2,1,1,
2,-1,2,-2,1,-2,-1


```

EXPLOSIE


Met behulp van dit programma speelt U het uitdagende spel Explosie tegen de computer. Durft U het aan ... ?!

Explosie wordt gespeeld op een bord van 3×3 of 4×4 velden. De computer vraagt U allereerst welke afmetingen (3×3 of 4×4) U wenst.


Ieder veld op het bord heeft een bepaalde capaciteit. De capaciteit is gelijk aan het aantal aangrenzende vakken. Dus een veld in de hoek heeft een capaciteit van 2.



Velden aan de rand hebben altijd een capaciteit van 3.



Velden in het midden hebben altijd een capaciteit van 4.



```

      5 REM EXPLOSIE
      10 GO TO 8000
     100 GO SUB 3000
     105 GO SUB 2000
     110 LET i=0: FOR x=1 TO k
     120 FOR y=1 TO k
     130 GO SUB 500
     140 IF a>b(x,y) THEN GO TO 170
     150 IF p1=1 THEN PRINT AT x0+xk
*x,y0+yk*y;" ";
     160 GO SUB 1000
     165 IF u=0 THEN RETURN
     170 NEXT y
     180 NEXT x
     190 IF i=1 THEN GO TO 110
     200 GO SUB 1500
     210 RETURN
     500 IF x<>1 AND x<>k AND y<>1 A
ND y<>k THEN LET a=4: RETURN
     510 IF x=y OR x+y=k+1 THEN LET
a=2: RETURN
     520 LET a=3: RETURN
    1000 IF y=k THEN GO TO 1040
    1010 LET b(x,y+1)=b(x,y+1)+1
    1020 LET d(x,y+1)=1
    1030 IF p1=1 THEN PRINT AT x0+xk
*x,y0+yk*(y+1); PAPER z(p);b(x,y
+1);
    1040 IF y=1 THEN GO TO 1080
    1050 LET b(x,y-1)=b(x,y-1)+1
    1060 LET d(x,y-1)=1
    1070 IF p1=1 THEN PRINT AT x0+xk
*x,y0+yk*(y-1); PAPER z(p);b(x,y
-1);
    1080 IF x=k THEN GO TO 1120
    1090 LET b(x+1,y)=b(x+1,y)+1
    1100 LET d(x+1,y)=1
    1110 IF p1=1 THEN PRINT AT x0+xk
*(x+1),y0+yk*y; PAPER z(p);b(x+1
,y);
    1120 IF x=1 THEN GO TO 1160
    1130 LET b(x-1,y)=b(x-1,y)+1
    1140 LET d(x-1,y)=1

```

```


1150 IF pl=1 THEN PRINT AT x0+xk
*(x-1),y0+yk*y; PAPER z(p);b(x-1
,y);
1160 LET i=1: LET b(x,y)=0
1170 LET e(x,y)=e(x,y)+1
1180 GO SUB 2300
1190 RETURN
1500 FOR m=1 TO k: FOR n=1 TO k
1510 IF e(m,n)<>0 OR d(m,n)=1 TH
EN LET c(m,n)=b(m,n)+1
1520 NEXT n: NEXT m
1530 RETURN
2000 FOR m=1 TO k: FOR n=1 TO k
2010 LET b(m,n)=ABS (c(m,n))
2020 NEXT n: NEXT m
2030 RETURN
2300 LET u=0
2310 FOR m=1 TO k: FOR n=1 TO k
2320 IF e(m,n)<3 THEN LET u=1: L
ET m=k: LET n=k
2330 NEXT n: NEXT m
2340 IF u=0 AND pl=1 THEN GO TO
9500
2350 RETURN
3000 FOR m=1 TO k: FOR n=1 TO k
3010 LET c(m,n)=f(m,n)
3020 NEXT n: NEXT m
3030 RETURN
4000 FOR m=1 TO k: FOR n=1 TO k
4010 LET f(m,n)=c(m,n)
4020 NEXT n: NEXT m
4030 RETURN
5000 LET s0=0
5010 FOR x=1 TO k: FOR y=1 TO k
5020 LET s0=s0+c(x,y): GO SUB 50
0
5030 IF -c(x,y)<>a-1 THEN GO TO
5170
5040 LET s0=s0-2
5050 IF x=k THEN GO TO 5080
5060 LET x=x+1: GO SUB 500: LET
x=x-1
5070 IF c(x+1,y)=a-1 THEN LET s0
=s0+10
5080 IF x=1 THEN GO TO 5110
5090 LET x=x-1: GO SUB 500: LET
x=x+1
5100 IF c(x-1,y)=a-1 THEN LET s0
=s0+10
5110 IF y=k THEN GO TO 5140
5120 LET y=y+1: GO SUB 500: LET
y=y-1
5130 IF c(x,y+1)=a-1 THEN LET s0
=s0+10
5140 IF y=1 THEN GO TO 5170
5150 LET y=y-1: GO SUB 500: LET
y=y+1
5160 IF c(x,y-1)=a-1 THEN LET s0
=s0+10

```

```

5170 NEXT y: NEXT x
5180 RETURN
7000 FOR m=1 TO k: FOR n=1 TO k
7010 LET d(m,n)=0: LET e(m,n)=0
7020 NEXT n: NEXT m
7030 RETURN
8010 PRINT AT 1,7;"*** EXPLODIE
***"
8020 DIM b(4,4): DIM c(4,4): DIM
d(4,4): DIM e(4,4): DIM f(4,4):
DIM z(2)
8030 LET k=4: GO SUB 7000
8040 FOR m=1 TO 4: FOR n=1 TO 4
8050 LET f(m,n)=0
8060 NEXT n: NEXT m
8070 LET x0=4: LET xk=2: LET y0=
11: LET yk=2
8080 LET xp=8*y0+12: LET yp=175-
(8*x0+12)
8090 LET z(1)=5: LET z(2)=6
8200 INPUT "WELK BORD KIEST U? (3
OF 4)."; k$;
8210 IF k$ <> "3" AND k$ <> "4" THEN
GO TO 8200
8220 LET k=VAL k$
8230 FOR i=0 TO k
8240 PLOT xp+16*i,yp: DRAW 0,-16
*k
8250 PLOT xp,yp-16*i: DRAW 16*k,
0
8260 NEXT i
8270 PLOT xp,yp: DRAW -12,12
8280 PRINT AT x0+1,y0;"X";
8290 PRINT AT x0,y0+1;"Y";
8300 FOR i=1 TO k
8310 PRINT AT x0+xk*i,y0;i;
8320 PRINT AT x0,y0+yk*i;i;
8330 NEXT i
8400 INPUT "WIE BEGINT? JIJ (J)
OF IK (I)?"; TAB 32; b$

```




```

8410 IF b$="J" OR b$="j" THEN GO
    TO 8600
8420 IF b$<>"I" AND b$<>"i" THEN
    GO TO 8400
8430 GO TO 8600
8600 PRINT AT 16,6;"NU IS HET JO
    UU BEURT! "; PRINT AT 18,0;TA
    B 31;""; PRINT AT 19,0;TAB 31;""
8610 INPUT "WAT IS JOUW ZET?";TA
    B 32;"X: ";g$
8620 IF g$<>"1" AND g$<>"2" AND
    g$<>"3" AND g$<>k$ THEN GO TO 86
    10
8630 PRINT AT 18,14;"X: ";g$;
8640 INPUT "WAT IS UW ZET?";TAB
    32;"Y: ";h$
8650 IF h$<>"1" AND h$<>"2" AND
    h$<>"3" AND h$<>k$ THEN GO TO 86
    40
8660 PRINT AT 19,14;"Y: ";h$;
8670 LET g=VAL g$; LET h=VAL h$
8680 IF f(g,h)<0 THEN BEEP 1,3:
    GO TO 8600
8690 LET f(g,h)=f(g,h)+1: LET t=
    1: LET p=1: LET p1=1
8700 GO SUB 9900
8710 GO SUB 100
8720 GO SUB 7000
8730 GO SUB 4000
8800 LET t=-1: LET p=2: LET p1=0
8810 PRINT AT 16,6;"NU IS HET MI
    JN BEURT! "; PRINT AT 18,0;TAB
    31;""; PRINT AT 19,0;TAB 31;""
8820 LET st=1000
8830 FOR q=1 TO k: FOR w=1 TO k
8840 IF f(q,w)>0 THEN GO TO 8930
8850 GO SUB 3000
8860 LET c(q,w)=c(q,w)-1
8870 GO SUB 105
8880 GO SUB 7000
8890 GO SUB 5000
8900 IF so<st THEN LET st=so: LE
    T v=q: LET z=w: GO TO 8930
8910 IF so<>st THEN GO TO 8930
8920 IF AND>.5 THEN LET st=so: L
    ET v=q: LET z=w
8930 NEXT w: NEXT q
8940 PRINT AT 18,14;"X: ";v;: PR
    INT AT 19,14;"Y: ";z;
8950 GO SUB 3000: BEEP 1,1: LET
    f(v,z)=f(v,z)-1
8960 LET g=v: LET h=z: GO SUB 99
    00
8970 LET p1=1: GO SUB 100
8980 GO SUB 7000
8990 GO SUB 4000
9000 GO TO 8600

```

```

9500 IF t=1 THEN INPUT "U HEEFT
GEWONNEN!";TAB 32;"NOG EEN KEER?
J OF N : ";a$: GO TO 9520
9510 INPUT "U HEEFT VERLOREN!";T
AB 32;"NOG EEN KEER? J OF N : ";
a$
9520 IF a$="j" OR a$="J" THEN RU
N
9530 STOP
9900 PRINT AT x0+xk*g,y0+yk*h; P
APER z(p);ABS f(g,h);
9910 RETURN

```

NEW YORK, NEW YORK

Bij dit geweldig spannende spel bevindt U zich hoog boven New York van waaruit U dan neerkijkt op al die straten. Op dit moment zijn de straten nog verlaten ... maar wacht, daar verschijnen van alle kanten auto's. Welnu, het is Uw taak om het verkeer zo goed mogelijk te laten doorstromen en wel door de verkeerslichten vanuit de helicopter te besturen. Als een auto (voorgesteld door een vierkantje) een rood licht bereikt, stopt hij natuurlijk. Elk stoplicht draagt een symbool - als U op de toets waarop dat symbool staat drukt, wordt het licht groen en de auto kan doorrijden.

Aan het begin van het spel heeft U 100 punten. U krijgt er 10 punten bij elke keer als een auto een kruispunt oversteekt en 25 punten voor elke auto die de gehele stad ongehinderd door kan rijden. Als echter twee auto's botsen ... ja dan verliest U 100 punten en de weg zal tijdelijk geblokkeerd zijn. Moet een auto stoppen voor een rood licht of voor een ongeluk dan kost U dat ook punten. Heeft U geen punten meer dan is het spel uit - zo niet dan geldt een vastgestelde tijd van ongeveer 5 minuten. Hoe meer punten U heeft - hoe beter.

Als U er ooit van droomde verkeersagent in die zo dynamische wereldstad te worden welnu, hier is Uw kans.

```
10 REM NEW YORK, NEW YORK
11 CLEAR 31500: LET sc=31500:
FOR i=sc TO sc+32*24: POKE i,0:
NEXT i
12 FOR i=1 TO 6: READ x: POKE
sc+x,255: NEXT i
13 DATA 75,84,354,381,619,628
20 DEF FN u()=INT ((65536*PEEK
23674+256*PEEK 23673+PEEK 23672
)/50): REM tijd in seconden
21 DEF FN m(x,y)=(x+y+ABS (x-y
))/2
22 DEF FN t()=FN m(FN u(),FN u
())
30 RANDOMIZE : DEF FN r(x)=INT
(RND*x)
70 LET z$="": LET ao=0: LET pn
=0: LET af=0: LET tl=1
80 LET aw=40: DIM x(40): DIM y
(40): DIM r(40): DIM l(42): DIM
k(42): DIM l$(42,1): DIM a(4): D
IM f(9): DIM g(9)
```

```


81 LET a(1)=1: LET a(2)=-1: LE
T a(3)=32: LET a(4)=-32
90 LET tb=FN t(): LET pu=100
99 REM schermlayout
100 BORDER 5: PAPER 0: CLS
110 INK 7: FOR i=1 TO 7
120 READ sx,sy,x,y: PLOT sx,sy:
DRAW x,y
130 NEXT i
140 DATA 0,147,255,0,28,83,199,
0,0,27,255,0
150 DATA 28,0,0,175,92,27,0,119
164,27,0,119,228,0,0,175
200 PAPER 2: FOR i=1 TO 42
210 READ l(i),k(i),l$(i): PRINT
AT l(i),k(i);l$(i)
220 NEXT i
230 DATA 1,2,"A",1,27,"B",2,5,"
C",2,13,"D",2,22,"E",2,30,"F",5,"
240 DATA 4,1,"G",4,9,"H",4,18,"
I",4,26,"J",5,4,"K",5,12,"L",5,2
1,"M",5,29,"N",
250 DATA 9,2,"O",9,10,"P",9,19,
"Q",9,27,"R",10,5,"S",10,13,"T",
10,22,"U",
260 DATA 12,9,"V",12,18,"W",12,
26,"X",13,4,"Y",13,12,"Z",13,21,
"0",13,29,"1",
270 DATA 16,2,"2",16,10,"3",16,
19,"4",16,27,"5",17,5,"6",17,13,
"7",17,22,"8",17,30,"9",
280 DATA 19,1,"!",19,9,"@",19,1
8,"#",19,26,"$",20,4,"%",20,29,"
&",
299 REM karakters definieren
300 FOR i=144 TO 147
310 FOR j=0 TO 7
320 READ x: POKE USA CHR$ i+j,x
330 NEXT j: NEXT i
340 DATA 170,85,170,85,170,85,1
70,85
350 DATA 0,0,0,0,255,0,0,0
360 DATA 8,8,8,8,8,8,8,8
370 DATA 8,8,8,8,255,8,8,8
500 PAPER 6: INK 0
510 PRINT AT 6,13;"Score:"
520 PRINT AT 20,11;"Tijd: 0:00"
530 PAPER 5
540 PRINT AT 0,6;"New York - N
ew York"
999 REM hoofdprogramma
1000 FOR i=1 TO aw: LET r=r(i):
LET a$=INKEY$: IF a$<>" " AND a$<
>z$ THEN GO SUB 3100
1002 LET z$=a$
1005 IF r=0 THEN GO TO 1099
1010 LET x=x(i): LET y=y(i): LET
sx=0: LET sy=0

```

```

1020 IF r=1 OR r=17 OR r=33 OR r
=65 THEN LET x=x+1: LET f=140: L
ET o0=131: LET o=145: LET sy=1:
IF x>31 THEN GO TO 1200
1030 IF r=2 OR r=18 OR r=34 OR r
=66 THEN LET x=x-1: LET f=131: L
ET o0=140: LET o=145: LET sy=-1:
IF x<0 THEN GO TO 1200
1040 IF r=4 OR r=20 OR r=36 OR r
=68 THEN LET y=y+1: LET f=138: L
ET o0=133: LET o=146: LET sx=-1:
IF y>21 THEN GO TO 1200
1050 IF r=8 OR r=24 OR r=40 OR r
=72 THEN LET y=y-1: LET f=133: L
ET o0=138: LET o=146: LET sx=1:
IF y<0 THEN GO TO 1200
1060 IF r>15 AND r<32 THEN GO TO
1100
1070 IF r>31 AND r<64 THEN GO TO
1400
1080 IF r>63 THEN LET o=147: LE
T r=r-64: LET r(i)=r: LET pu=pu+
10
1090 GO TO 1300
1099 NEXT i: GO TO 2000
1100 IF (ATTR (y(i)+sy,x(i)+sx))
>=32 THEN LET r=r+16: LET r(i)=r
: GO TO 1300
1110 LET pu=pu-1: GO TO 1099
1200 LET pu=pu+25: LET pn=pn+1:
LET r(i)=0: GO TO 1333
1300 LET a=PEEK (sc+32*y+x): IF
a=f OR a=143 THEN LET pu=pu-1: G
O TO 1099
1310 IF SCREEN$ (y+sy,x+sx)<>" "
THEN LET r(i)=r+16
1320 IF a=00 THEN LET f=143
1325 LET a=PEEK (sc+32*y(i)+x(i)
): IF a=143 THEN LET o=00
1327 IF a=144 THEN LET r(i)=0: L
ET ao=ao+1: LET pu=pu-100: GO TO
1099
1330 PAPER 0: INK 6: PRINT AT y,
x:CHR$ f: POKE sc+32*y+x,f
1333 PAPER 0: INK 7: PRINT AT y(
i),x(i):CHR$ o: POKE sc+32*y(i)+
x(i),o: LET x(i)=x: LET y(i)=y
1340 GO TO 1099
1400 LET a=PEEK (sc+32*y+x): IF
a=f THEN GO TO 1099
1410 IF a<>147 AND a<>0 THEN GO
TO 1500
1420 FOR k=0 TO 2: LET j=FN r(4)
: LET b=PEEK (sc+32*y+x+a(j+1)):
IF NOT (b=145 OR b=146 OR b=0)
THEN NEXT k: GO TO 1099
1430 LET k=2: NEXT k: LET r(i)=2
+1+64: GO TO 1330

```



```

1500 LET ao=ao+1: LET r(i)=0: LE
T pu=pu-100: IF a<>144 THEN BEEP
  2,-12: LET af=af+1: LET f(af)=x
: LET g(af)=y
1510 LET f=144: GO TO 1330
2000 IF r(tl)=0 THEN GO TO 2100
2010 LET tl=tl+1: IF tl>aw THEN
LET tl=1
2020 GO TO 3000
2100 LET a=FN r(4): LET r(tl)=2↑
a: GO TO 2110+10*a
2110 LET f=140: LET x(tl)=0: GO
TO 2200
2120 LET f=131: LET x(tl)=31: GO
TO 2200
2130 LET f=138: LET y(tl)=0: GO
TO 2300
2140 LET f=133: LET y(tl)=21: GO
TO 2300
2200 IF FN r(2)=1 THEN LET y(tl)
=3: GO TO 2400
2210 LET y(tl)=18: GO TO 2400
2300 IF FN r(2)=1 THEN LET x(tl)
=3: GO TO 2400
2310 LET x(tl)=28: GO TO 2400
2400 PAPER 0: INK 6: PRINT AT y(
tl),x(tl);CHR$ f
3000 LET a#=INKEY$: IF a#<>"" AN
D a#<>z$ THEN GO SUB 3100
3002 LET z#=a$
3010 GO TO 4000
3100 LET a=CODE a$
3110 IF a>64 AND a<91 THEN LET b
=a-64: GO TO 3200
3111 IF a>96 AND a<123 THEN LET
b=a-96: GO TO 3200
3120 IF a>47 AND a<58 THEN LET b
=a-21: GO TO 3200

```

```

3130 IF a>32 AND a<39 THEN LET b
=a+4: GO TO 3200
3140 IF a=64 THEN LET b=38: GO T
O 3200
3150 RETURN
3200 BEEP .1,12
3210 IF (ATTR (L(b),k(b))<32) TH
EN LET a=4: GO TO 3300
3220 LET a=2
3300 PAPER a: INK 9: PRINT AT (
b),k(b);l$(b)
3310 RETURN
4000 PAPER 6: INK 0
4010 LET t=FN t()-tb: LET tm=INT
(t/60): PRINT AT 20,17;tm;":":(
STR$ (100+t-tm*60))(2 TO 3)
4020 IF pu<0 THEN LET pu=0
4030 LET a$=CHR$ 143+STR$ (pu):
PRINT AT 8,18-LEN (a$);a$
4040 IF pu=0 THEN GO TO 9000
4100 IF FN r(25) OR af=0 THEN GO
TO 4200
4110 PAPER 0: INK 7: PRINT AT g(
1),f(1);CHR$ 147: POKE sc+32*g(1
)+f(1),147: FOR i=1 TO 8: LET f(
i)=f(i+1): LET g(i)=g(i+1): NEXT
i: LET af=af-1
4200 IF FN t()-tb>300 THEN GO TO
9000
5000 GO TO 1000
9000 BEEP 5,-12: INPUT "Nog een
keer? ";a$
9010 IF a$(1)="n" THEN GO TO 910
0
9020 RUN
9100 PAPER 7: INK 0: CLS

```


CODE KRAKEN

Het zou hier om een spannend spionageverhaal kunnen gaan waarbij we zo snel mogelijk een brandkast moeten kraken om aan belangrijke informatie te komen. De code van de brandkast kent U natuurlijk nog niet en het is de vraag in hoeveel beurten U die kunt vinden. Doe er niet te lang over ... stel dat er iemand komt!

De code is opgeslagen in een computer en we weten dat die code bestaat uit drie series van 20 enen en nullen, waarbij we steeds uitgaan van de volgende serie:

01101010111010100101

Die drie series zijn door „cyclische verwisseling” verkregen, in gewoon Nederlands wil dat zeggen dat getallen van het linkeruiteinde naar rechts worden verplaatst. Hier zien we een voorbeeld van 3 van dergelijke series tesamen met de som van die getallen voor iedere kolom.

RIJ 1 →	00101011101010110101
RIJ 2 →	01110101011010100101
RIJ 3 →	01001010111010101101
SOM →	02212122223030311303

Deze series zijn natuurlijk voor U verborgen. Sterker nog U moet deze series weer zien terug te vinden. Als het spel begint krijgt U ook drie rijen getallen te zien, deze zijn ook verkregen door cyclische verwisseling van de reeks ... helaas deze verwisseling is willekeurig.

Een voorbeeld: als de bovenste rij 2 posities is opgeschoven, de middelste 4 en de onderste 8, dan zien we op het beeldscherm:

1	0	1	0	1	1	1	0	1	0	1	0	1	1	0	1	0	1	0	0
0	1	0	1	0	1	1	0	1	0	1	0	0	1	0	1	0	1	1	1
1	1	1	0	1	0	1	0	1	1	0	1	0	1	0	0	1	0	1	0
<hr/>																			
2	0	0	0	0	1	1	2	1	1	1	1	2	3	3	1	0	1	2	2

Merk op dat we nu niet de som van iedere nieuwe kolom te zien krijgen, maar juist het verschil tussen deze som en de eerste (verborgen) som. Voor de eerste kolom geldt bijvoorbeeld dat de verborgen som 0 is en aangezien de som op het beeldscherm 2 is krijgen we nu $2 - 0 = 2$ te zien. Bij de tweede kolom zijn zowel de verborgen som als de nieuwe som 2 dus we zien $2 - 2 = 0$. We moeten nu steeds de rijen verschuiven en wel tot ze precies gelijk zijn aan de verborgen series. In dit geval zijn alle verschillen natuurlijk 0, kortom de onderste serie op het beeldscherm toont dan alleen maar nullen. De bovenste rij heeft nummer 1, de middelste 2 en de onderste nummer 3. Als we bijvoorbeeld aangeven:


RIJ: 2
STAPPEN: 1

Dan zal de middelste rij een positie naar links opschuiven. Hoeveel beurten heeft U nodig om de code te kraken? We kennen super-spionnen die de klus in 10 beurten klaren

```

1  REM CODE KRAKEN
5  LET nt=0
10 DIM t(4,20)
20 LET a$="1010010101110101011
0"
30 LET flag=0
40 RANDOMIZE
50 FOR j=1 TO 3
60 LET r=INT (RND*19+1)
70 FOR k=1 TO 20
80 LET dis=k+r
90 IF dis<21 THEN GO TO 110
100 LET dis=dis-20
110 LET t(j,k)=VAL a$(dis)

```



```

120 NEXT k
130 NEXT j
140 IF flag>0 THEN GO TO 180
150 FOR k=1 TO 20
160 LET t(4,k)=t(1,k)+t(2,k)+t(
3,k)
170 NEXT k
180 LET flag=flag+1
190 IF flag>1 THEN GO TO 210
200 IF flag=1 THEN GO TO 50
210 CLS
220 LET tel=0
230 FOR k=1 TO 20
240 PRINT AT 10,k+2;t(1,k)
250 PRINT AT 11,k+2;t(2,k)
260 PRINT AT 12,k+2;t(3,k)
270 LET v=t(4,k)-t(3,k)-t(2,k)-
t(1,k)
280 LET v=ABS (v)
290 IF v=0 THEN LET tel=tel+1
310 PRINT AT 14,k+2;v
320 NEXT k
330 IF tel=20 THEN GO TO 450
340 INPUT "RIJ=";j
350 INPUT "STAPPEN=";s
360 LET nt=nt+1
370 FOR i=1 TO s
380 LET h=t(j,1)
390 FOR k=1 TO 19
400 LET t(j,k)=t(j,k+1)
410 NEXT k
420 LET t(j,20)=h
430 NEXT i
440 GO TO 210
450 PRINT FLASH 1;"Gevonden in
";nt;" beurten"


```

BLACKBOX


Slechts gewapend met een laserstraal moet U bij dit spel trachten te ontdekken waar zich binnen een Blackbox atomen bevinden.

Deze Blackbox bestaat uit $8 \times 8 \times 8$ kubusjes. De atomen kunnen in elke kubus verborgen zijn behalve in de kubusjes van de buitenste laag. Er zijn altijd 5 atomen in de Blackbox verborgen. De atomen zullen de laserstralen steeds reflecteren of afbuigen en wel volgens de volgende regels:

- Een laserstraal die een atoom raakt wordt weer in de richting van waaruit hij kwam gereflecteerd.




- Een straal die door een kubus grenzend aan een kubus met een atoom gaat, zal onder een rechte hoek worden teruggekaatst.




- Een straal die gericht wordt op een kubus waar in diagonale richting een kubusje met een atoom aan grenst, zal in een rich-

ting gereflecteerd worden die verkregen wordt door de optelling van 2 rechthoekige reflecties.



De Blackbox is genummerd als volgt:



De computer zal allereerst om de coördinaten P1, P2 en P3 vragen. Dit zijn de coördinaten van het kubusje van waaruit U de laserstraal de Blackbox wilt laten binnengaan.

Het is duidelijk dat deze coördinaten steeds een kubusje van de schil moeten aangeven, m.a.w. minstens één van de coördinaten moet 1 of 8 zijn.

Als U eenmaal een schot heeft ingevoerd, zal de computer vertellen in welk punt de straal de kubus weer verlaten heeft. Bedenk dat één straal door meer dan één atoom gereflecteerd kan worden.

Probeer in een zo gering mogelijk aantal beurten uit te vinden waar de atomen zich bevinden ... maar we waarschuwen, gemakkelijk is het zeker niet!

```
10 REM BLACKBOX
20 DIM b(10,10,10)
30 GO TO 1000
100 REM schot invoeren
110 PRINT "Schot invoeren"
120 GO SUB 500: REM coördinaten
invoeren
130 LET dx=(px=2)-(px=9)
140 LET dy=(py=2)-(py=9)
150 LET dz=(pz=2)-(pz=9)
160 IF (dx=0) AND (dy=0) AND (d
z=0) THEN GO TO 120
170 RETURN
200 REM resultaat berekenen
210 FOR x=-1 TO 1
220 FOR y=-1 TO 1
230 FOR z=-1 TO 1
240 IF b(px+x,py+y,pz+z) <> 1 THE
N GO TO 260
250 LET dx=dx-x
260 LET dy=dy-y
270 LET dz=dz-z
280 NEXT z
290 NEXT y
300 NEXT x
310 LET dx=SGN dx: LET dy=SGN d
y: LET dz=SGN dz
320 LET px=px+dx: LET py=py+dy:
LET pz=pz+dz
330 IF (px=1)+(px=10)+(py=1)+(p
y=10)+(pz=1)+(pz=10)=0 THEN GO T
O 210
340 PRINT "Resultaat : ";px-dx
-1;" ";py-dy-1;" ";pz-dz-1
350 RETURN
```

```

400 REM gissing invoeren
410 PRINT "Gissing invoeren"
420 GO SUB 500
430 IF b(px,py,pz)=1 THEN PRINT
"Goe!" : RETURN
440 PRINT "Fout!"
450 RETURN
500 REM coördinaten invoeren
510 PRINT "Coördinaten: ";
520 INPUT "P1=";px: LET px=px+1
530 IF px<2 OR px>9 THEN GO TO
520
535 PRINT px-1;
540 INPUT "P2=";py: LET py=py+1
550 IF py<2 OR py>9 THEN GO TO
540
555 PRINT " ";py-1;
560 INPUT "P3=";pz: LET pz=pz+1
570 IF pz<2 OR pz>9 THEN GO TO
560
580 PRINT " ";pz-1
590 RETURN
1000 REM hoofdprogramma
1010 FOR a=1 TO 5
1020 LET px=INT (RND*6+3): LET p
y=INT (RND*6+3): LET pz=INT (RND
*6+3)
1025 IF b(px,py,pz)=1 THEN GO TO
1020
1030 LET b(px,py,pz)=1
1035 NEXT a
1040 GO SUB 100: REM schot invoe
ren
1050 GO SUB 200: REM resultaat
berekenen
1060 INPUT "Schieten of raden ?
(s of r) ";t$
1070 IF t$<>"s" AND t$<>"r" THEN
GO TO 1060
1080 IF t$="s" THEN GO TO 1040
1090 GO SUB 400: REM gissing inv
oeren
1100 GO TO 1060

```



SCHATZOEKEN

Een verre reis naar een verborgen schat ... wie zou dat niet willen! We gaan over woeste bergen, lieflijke dalen, onbegaanbare wegen en dorre woestijnen om dan uiteindelijk de verboden plaatsen te bereiken waar de glinsterende schatten verborgen zijn.

Lijkt U dat wat ... ja ... dan is dit echt een spel voor U!

U wordt hoog in de Rocky Mountains neergezet en U krijgt een kaart waarop al de verborgen schatten staan en tevens wat ze waard zijn.

Om die schatten te bemachtigen moet U over smalle slingerende, kortom gevaarlijke paden gaan, één verkeerde stap en een zekere dood wacht U.



We bepalen onze richting door middel van de cursorbesturingstoetsen en we bemachtigen de schatten door er simpelweg op te gaan staan. Maar wacht ... zo eenvoudig is het niet. U dacht toch niet werkelijk dat er niet nog ergens gevaar dreigde?! De schatten die U zoekt worden helaas zorgvuldig bewaakt en wel door zeer gevaarlijke angstaanjagende Kamikaze-vogels die U aanvallen. Als ze U in de gaten krijgen raken ze buiten zichzelf van woede. De enige manier om ze te stoppen is door op de 1 toets te drukken.

Durft U de uitdaging nog aan te nemen ...? Oké, begin in de hoek linksonder en verzamel zoveel schatten als U kunt, breng ze in veiligheid door de uitgang rechtsboven. Dit levert bonuspunten op en bovendien de kans om meer schatten te vergaren.

Helaas zult U merken dat de Kamikaze-vogels steeds aggressiever worden.

Grote risico's dus, niettemin de moeite waard. Dus ... verzamel Uw moed en ... sterkte!

```

5 REM SCHAT ZOEKEN
10 DIM b$(15,25)
20 LET nl=3: LET br=0
30 LET nc=0: LET p1=0: LET p2=
0
50 LET u$="11"
60 DATA "
70 DATA " 3 2
80 DATA " 5 4 6
90 DATA " 3 5 7
100 DATA " 3 5 9
110 DATA " 3 5 9
120 DATA " 3 4 8
130 DATA " 1 3 3
140 DATA " 1 1 3
150 DATA " 1 3 7
160 DATA "
170 DATA "
180 DATA "
190 DATA "
200 DATA "
210 GO TO 1000
300 REM **schermlayout**
310 RESTORE : CLS
320 FOR y=1 TO 15
330 READ b$(y)
340 PRINT AT y+2,6;b$(y)
350 NEXT y
360 LET x=2: LET y=14
370 PRINT AT y+2,x+5;"*"
390 PRINT AT 19,6;"RONDE: ";nc)
AT 19,15;"GULDENS: ";p2
400 RETURN
450 REM **speler beweegt**
460 LET ti=3+RND*(100-10*nc)
470 FOR u=1 TO ti
480 LET t$=INKEY$
490 IF t$="" THEN GO TO 480
500 LET dx=(t$="8")-(t$="5")
510 LET dy=(t$="6")-(t$="7")
520 LET s#=b$(y+dy,x+dx)
530 IF s$="■" THEN GO TO 500
540 PRINT AT y+2,x+5)" "

```

```

550 LET X=X+dx: LET Y=Y+dy
555 LET p0=CODE S$-48
560 IF p0<0 THEN GO TO 600
570 LET p1=p1+p0: LET p2=p2+p0
580 PRINT AT 19,24;P2
590 LET b$(Y,X)=" "
600 PRINT AT Y+2,X+5;"#"
610 IF X=19 AND Y=1 THEN LET U=
ti: LET br=1
620 NEXT U
630 RETURN
700 REM **vogel valt aan**
710 LET t$=INKEY$
720 LET dt=15-3*nc: LET x5=x+5
730 FOR b=29 TO x5 STEP -1
740 PRINT AT Y+2,b;U$
750 FOR e=1 TO dt: NEXT e

```



```

760 IF INKEY$="1" THEN RETURN
770 NEXT b
780 RETURN
1000 REM **hoofdprogramma
1010 GO SUB 300: REM scherm layou
t
1020 GO SUB 450: REM speler bewe
egt
1030 IF br=1 THEN LET br=0: LET
nc=nc+1: LET p2=p2+nc*p1: LET p1
=0: GO TO 1010

```

```

1040 GO SUB 700: REM vogel
1050 IF b<x+5 THEN GO TO 1100
1060 PRINT AT y+2,6;b$(y)
1070 PRINT AT y+2,x+5;"*"
1080 LET p1=p1+1
1090 GO TO 1020
1100 BEEP .5,2: CLS
1105 IF p2<>1 THEN LET Q$="S"
1107 IF p2=1 THEN LET Q$=""
1110 PRINT "U HEEFT ";p2; " GULD
EN";Q$;" GEVONDEN"

```

I.T. - DE ADVENTURE VAN DE EEUW

In deze spannende adventure (avontuur) gaat U een kijkje nemen in de duistere ingewanden der aarde. U zult ongetwijfeld het vreemde wezen dat daar leeft ontmoeten. Zijn naam is natuurlijk I.T. wat zoveel betekent als Intra Terrestrial.

Als U nog Nooit eerder een adventure heeft gespeeld hoeft U zich geen zorgen te maken over de regels, die zijn er namelijk niet! Kortom U moet alles zelf uitzoeken. U reist beneden het oppervlak van moedertje aarde, U zult moeilijke en ogenschijnlijk vreemde taken moeten uitvoeren en grote gevaren onder ogen durven zien ... het gaat tenslotte om een avontuur. Wordt dit alles U teveel type dan „help” in en kijk wat de computer U te zeggen heeft.

```
10 REM I.T.
100 GO SUB 5000
110 LET ihd=8: LET nkam=0
120 GO TO 140
130 GO SUB 500
140 GO SUB 1000
150 GO SUB 800
160 IF cwd<5 THEN GO TO 130
170 CLS
180 PRINT "Jammer, u heeft uzelf
gedood"
200 STOP
500 REM invoer*****
**
510 LET nkam=0
520 LET ihd=0
530 INPUT "Wat wilt u doen :";c$
$
550 LET lng=LEN (c$)
560 FOR i=1 TO 15
570 IF h$(i, TO lng)=c$ THEN LET
T ihd=i: LET i=16
580 NEXT i
590 IF ihd>8 THEN RETURN
600 IF ihd=0 THEN GO TO 650
610 LET nkam=k(ikm,ihd)
620 IF nkam<>0 THEN RETURN
630 PRINT "Die kant kunt u niet
op"
640 GO TO 500
650 PRINT "Ik begrijp u niet"
660 GO TO 500
670 RETURN
```

```

800 REM uitvoer
810 PRINT "u bent in de ";r$(ik
m)
811 IF ikm>24 OR bll=1 THEN GO
TO 820
812 PRINT "U ziet niet veel"
813 GO TO 830
820 IF ivw<>0 THEN PRINT "Hier
is ";v$(ivw)
830 FOR i=1 TO 6
840 IF k(ikm,i)<>0 THEN PRINT "
U kunt naar ";h$(i)
850 NEXT i
860 PRINT
870 PRINT "          ***"
880 PRINT
890 RETURN
1000 REM handelingen
1010 IF nka<>0 THEN LET ikm=nka
m
1020 LET ivw=0
1030 FOR i=1 TO 12
1040 IF v(i)=ikm THEN LET ivw=i:
LET i=12
1050 NEXT i
1060 LET ips=0
1070 FOR i=1 TO 6
1080 IF p(i,1)=ikm THEN LET ips=
i: LET i=6
1090 NEXT i
1100 REM algemene handelingen***
**
1110 IF (ikm<12) AND (cri<>8) TH
EN GO SUB 4560
1120 IF bll=1 THEN GO SUB 4700
1200 REM spelers handelingen***
1210 IF ihd<7 THEN GO TO 1240
1220 GO SUB 1300+(ihd-7)*200
1230 REM handlers
1240 IF ips=0 THEN RETURN
1250 GO SUB 3000+(ips-1)*200
1260 RETURN
1300 REM **doden**
1310 IF (ips=0) OR (ips>4) THEN
PRINT "Wilt u de tijd doden of z
o?": RETURN
1320 LET ivwb=7
1330 GO SUB 4800
1340 IF ibz=0 THEN PRINT "Niet a
anvallen zonder wapen, dat is zel
fmoord!": RETURN
1350 PRINT "U tracht de ";p$(ips
);" te doden"
1360 PRINT "U valt hem aan met u
w zwaard,"
1370 IF (RND<.5) AND (p(ips,2)<4
) THEN PRINT "maar hij ontwijkt
uw slag.": GO TO 1305

```

```

1380 PRINT "en u geeft hem een f
linke klap"
1390 LET p(ips,3)=p(ips,3)-1
1395 IF RND>.2 THEN GO TO 1440
1400 PRINT "Dit is uw kans! U ku
nt toeslaan voordat hij zich her
stelt!"
1410 INPUT "Doet u dat? ";y$
1420 IF y$<>"j" THEN GO TO 1440
1430 PRINT "U heft opnieuw uw zw
aard,"; GO TO 1370
1440 IF p(ips,3)>1 THEN RETURN
1450 IF p(ips,3)=1 THEN PRINT "U
w slachtoffer is zwaargewond,";
RETURN
1460 PRINT "U heeft hem gedood"
1470 LET p(ips,1)=0: LET ivw=8+i
ps
1480 LET v(ivw)=ikm: LET ips=0
1490 RETURN
1500 REM **help**
1510 PRINT "Mogelijke kommando's
:"
1520 FOR i=1 TO 15
1530 PRINT h$(i)
1540 NEXT i
1550 RETURN
1700 REM **meenemen**
1710 IF ivw<>0 THEN GO TO 1740
1720 PRINT "Er valt hier niets m
ee te nemen"
1730 RETURN
1740 PRINT "U pakt ";v$(ivw)
1750 LET b(ibmax)=ivw
1760 LET ibmax=ibmax+1
1770 LET v(ivw)=0
1780 LET ivw=0
1790 RETURN
1900 REM **neerleggen**
1910 INPUT "Wat wilt u neerlegge
n? ";d$
1915 LET lng=LEN (d$)
1920 LET ivwb=0
1930 FOR i=1 TO 12
1940 IF d$=v$(i, TO lng) THEN LE
T ivwb=i: LET i=12
1950 NEXT i
1960 IF ivwb<>0 THEN GO TO 1990
1970 PRINT "Ik kan u niet volgen
"
1980 RETURN
1990 GO SUB 4800
2000 IF ibz<>0 THEN GO TO 2030
2010 PRINT "Dat hebt u niet eens
"
2020 RETURN
2030 PRINT "U legt ";v$(b(ibz));
" neer"
2040 LET ivw=b(ibz)

```




```

2050 LET v(b(ibz))=ikm
2060 LET b(ibz)=b(ibmax-1)
2070 LET ibmax=ibmax-1
2075 IF ikm<>25 THEN RETURN
2080 INPUT "Wilt u nog iets neer
leggen? ";y$
2085 IF y$="j" THEN GO TO 1910
2090 RETURN
2100 REM **status**
2110 IF ibmax<>1 THEN GO TO 2140
2120 PRINT "U bezit slechts uw k
leren"
2130 GO TO 2180
2140 PRINT "U heeft:"
2150 FOR i=1 TO ibmax-1
2160 PRINT v$(b(i))
2170 NEXT i
2180 PRINT "U kunt nog ";5-cwd;"
wonden overleven"
2190 RETURN
2300 REM **verband**
2310 LET ivwb=6
2320 GO SUB 4800
2330 IF ibz<>0 THEN GO TO 2360
2340 PRINT "Dat speelt u zonder
verband nietklaar"
2350 RETURN
2360 PRINT "Dat ziet er niet bes
t uit! U heeft de hele voorra
ad nodig"

```

```


2380 BEEP 1,1
2390 PRINT "Goed, het zal wel ev
en houden"
2400 LET cwd=0
2410 LET b(ibz)=b(ibmax-1)
2420 LET ibmax=ibmax-1
2430 RETURN
2500 REM **lantaren aan**
2510 LET ivwb=1
2520 GO SUB 4800
2530 IF ibz<>0 THEN GO TO 2560
2540 PRINT "En waar mag die lant
aren wezen?"
2550 RETURN
2560 PRINT "De lantaren is aan"
2570 LET bll=c11>1
2580 RETURN
2700 REM **lantaren uit**
2710 LET ivwb=1
2720 GO SUB 4800
2730 IF ibz<>0 THEN GO TO 2760
2740 PRINT "U heeft er niet eens
een"
2750 RETURN
2760 PRINT "De lantaren is uit"
2770 LET bll=0
2780 RETURN
2900 REM **lezen**
2905 LET ivwb=4
2910 GO SUB 4800
2915 IF ibz<>0 THEN GO TO 2930
2920 PRINT "Wat wilt u lezen zon
der boek?"
2925 RETURN
2930 IF bll=1 OR ikm>24 THEN GO
TO 2960
2940 PRINT "Het is te donker om
te lezen"
2950 RETURN
2960 PRINT "Er staat een koekies
recept in het boek"
2970 PRINT "Er staat: Neem een h
ellehond en een snufje meel,
en leg dat op een barbecue"
2980 PRINT "Dat is alles"
2990 RETURN
3000 REM **I.T.**
3010 IF ivw=3 THEN LET p(1,2)=3:
LET v(ivw)=0: LET ivw=0
3020 GO TO 3020+p(1,2)*10
3030 PRINT "I.T. (de intraterres
trial) is hier. Hij praat opge
wonden, maar u verstaat er geen w
oord van"
3033 LET p(1,2)=2
3034 RETURN
3040 PRINT "I.T. gebaart dat hij
enorme dorst heeft. Hij kij
kt wanhopig"

```

```

3043 RETURN
3050 PRINT "I.T. drinkt lang en
diep. Na eenpoosje begint hij we
er te pratenen nu verstaat u hem
!"
3054 PRINT "Hij zegt: ""Help als
tublieft! Ga in de grotten op zo
ek naar de nullity-bom"
3055 PRINT """"Een gekke professo
r wil er de aarde mee opblazen"
3056 PRINT """"Ik heb gevochten m
et het waak-monster, maar nu be
n ik uitgeput dus u moet de aarde
redden!""""
3058 PRINT "Plotseling klapt hij
in elkaar."
3059 LET p(1,2)=4: RETURN
3060 PRINT "Het ziet er naar uit
dat I.T. in coma ligt"
3061 PRINT "Nu moet u het alleen
doen!"
3062 PRINT "VEEL SUCCES!"
3063 LET p(1,2)=5
3064 RETURN
3070 PRINT "I.T. is hier"
3071 PRINT "Hij ligt in coma"
3072 RETURN
3200 REM **koekiemonster**
3210 IF ivw=8 THEN LET p(2,2)=5:
LET v(ivw)=0: LET ivw=0

```



```

3220 GO TO 3220+P(2,2)*10
3230 PRINT "Hier is een gigantisch monster. Het rolt met zijn ogen en brult: ""Ben jij een koekie?!" ""
3233 LET p(2,2)=2
3234 RETURN
3240 PRINT "Het monster brult harder en harder: ""BEN JIJ EEN KOEKIE!!?!""
3242 LET p(2,2)=3
3243 RETURN
3250 PRINT "Het monster blijft brullen, en wordt behoorlijk agressief"
3252 LET p(2,2)=4
3253 RETURN
3260 PRINT "Het monster geeft u een vrese-lijke klap. Uw hoofd telt."
3262 LET cwd=cwd+1
3263 LET p(2,2)=INT (RND*3)+2
3264 RETURN
3270 PRINT "Het monster zegt verast: ""Koekies?""
3271 PRINT "Het begint direkt te eten"
3272 PRINT "Een enorme !!DREUN!! en het valt in slaap.": BEEP .2, 1
3273 LET k(16,6)=1
3274 LET p(2,2)=6
3275 RETURN
3280 PRINT "Het monster slaapt"
3281 RETURN
3400 REM **slang**
3410 PRINT "Er is hier een glibberige slang"
3420 IF RND<.4 THEN RETURN
3430 IF (ibmax=1) OR (RND<.5) THEN GO TO 3470
3440 LET ibmax=ibmax-1
3450 LET v(b(ibmax))=13+INT (RND*12)
3460 PRINT "Met een vlugge beweging pikt hij iets, en glibbert weg"
3480 LET p(3,1)=p(3,1)+3
3490 IF p(3,1)>24 THEN LET p(3,1)=p(3,1)-8
3500 RETURN
3600 REM **hellehond**
3620 GO TO 3620+P(4,2)*10
3630 PRINT "U ziet een reusachtige hellehond die kijkt alsof hij u wel zou lusten"
3632 LET p(4,2)=2
3633 RETURN

```

```

3640 PRINT "De hellehond valt aa
n en bijt"
3641 LET cwd=cwd+1
3642 LET p(4,2)=3
3643 RETURN
3650 PRINT "De hellehond gromt e
n lijkt een nieuwe aanval voor t
e bereiden"
3651 LET p(4,2)=2+INT (RND*2)
3652 RETURN
3800 REM **bom**
3810 PRINT "De nullity bom is hi
er"
3820 PRINT "er zijn drie draden
tussen de bom en de tijd klok:
blauw(b), geel(y) en rood(r)"
3825 PRINT "Twee ervan moet u lo
smaken om de explosie te voorkome
n."
3830 INPUT "Welke is de eerste?
";x$
3840 INPUT "En de tweede? ";y$
3850 LET cbo=0: LET c$="grb"
3860 FOR i=1 TO 3
3870 IF x$=c$(i) THEN LET cbo=cbo+1
3880 IF y$=c$(i) THEN LET cbo=cbo+1
3890 NEXT i
3900 IF cbo<2 THEN PRINT "Pas op
! Verkeerde invoer!": GO TO 3820
3910 IF (CODE (x$)*CODE (y$)=111
72) THEN GO TO 3950
3920 CLS
3930 PRINT "Enorme explosie! Pad
destoetwolk!Helaas...uw dappere
poging heeft de aarde niet kunnen
redden.."
3940 STOP
3950 FLASH 1: CLS
3960 PRINT "Gefeliciteerd! U ben
t geslaagd waar velen faalden!"
3970 STOP
4000 REM **barbecue**
4010 IF ivw=0 THEN GO TO 4060
4020 LET p(6,2)=2
4030 IF (v(2)=ikm) AND (v(12)=ik
m) THEN LET ivw=8: LET v(ivw)=ik
m: LET v(2)=0: LET v(12)=0: GO T
O 4060
4040 LET v(ivw)=0
4050 LET ivw=0
4060 GO TO 4060+10*p(6,2)
4070 PRINT "Hier is een enorme b
arbecue, meteen flink vuur erond
er"
4072 RETURN

```



```
4080 PRINT "Een felle flits! Plo  
tseling ver-spreidt zich een doo  
rdringende stank."  
4082 LET p(6,2)=3  
4083 RETURN  
4090 PRINT "Alles is weer rustig  
; Zelfs de stank is opgetrokken  
"  
4092 LET p(6,2)=1  
4093 RETURN  
4700 REM **licht**  
4710 LET cll=cll-bll  
4720 IF cll>5 THEN RETURN  
4730 IF cll<>5 THEN GO TO 4760  
4740 PRINT "Uw lantaren begint t  
e flakkeren": BEEP .2,1  
4750 RETURN  
4760 IF cll<>0 THEN RETURN  
4770 PRINT "Uw lantaren geeft de  
geest": BEEP .5,1  
4780 LET bll=0  
4790 RETURN  
4800 REM **bezittingen"  
4810 LET ibz=0  
4820 FOR i=1 TO ibmax-1  
4830 IF b(i)=ivwb THEN LET ibz=i  
: LET i=ibmax-1  
4840 NEXT i  
4850 RETURN
```

```

4860 REM **puzzel**
4865 LET cro=cro+1
4870 IF ikm=d(cro) THEN LET cri=
cri+1
4875 IF cro<8 THEN RETURN
4880 IF cri=8 THEN GO TO 4915
4885 PRINT "Er gebeurt iets vree
mds. Het is opeens stikdonker en
het lijkt of u opgetild wordt.
..Even bent u buiten bewustzijn.
4900 LET cro=1: LET cri=1
4905 LET ikm=1
4910 RETURN
4915 PRINT "U hoort een vreemd g
eluid, alsofer iets verschoven w
ordt"
4920 PRINT "Dan wordt het weer s
til."
4925 LET k(2,1)=3
4930 RETURN
5000 REM init*****
5010 DIM k(36,6): DIM r$(36,15)
5020 DIM h$(15,15)
5030 DIM p(6,3): DIM p$(6,10)
5040 DIM v(12): DIM v$(12,12)
5050 DIM b(12)
5060 DIM d(8)
5100 REM connections
5110 FOR i=1 TO 36
5120 LET k(i,1)=i+1: LET k(i,2)=
i-1
5130 LET k(i,3)=i+4: LET k(i,4)=
i-4
5140 NEXT i
5150 FOR i=0 TO 24 STEP 12
5160 FOR j=1 TO 9 STEP 4
5170 LET k(i+j+3,1)=0: LET k(i+j
,2)=0
5180 NEXT j
5190 FOR j=1 TO 4
5200 LET k(i+j+8,3)=0: LET k(i+j
,4)=0
5210 NEXT j
5220 NEXT i
5230 LET k(1,5)=16: LET k(7,5)=1
5
5240 LET k(32,6)=13: LET k(13,5)
=32
5250 LET k(35,6)=18: LET k(18,5)
=35
5260 FOR i=1 TO 15
5270 READ ikm: READ ihd
5280 LET k(ikm,ihd)=0
5290 NEXT i
5300 REM init parameters
5310 LET ikm=36
5320 LET ibz=1: LET ibmax=1
5330 LET ckm=0

```



```

5340 LET cll=100
5350 LET bll=0
5360 LET cwd=0
5370 LET cro=0: LET cri=0
5400 REM arrays inlezen
5410 FOR i=1 TO 15
5420 READ h$(i)
5430 NEXT i
5440 FOR i=1 TO 12
5450 READ v$(i)
5460 NEXT i
5470 FOR i=1 TO 6
5480 READ p$(i)
5490 NEXT i
5500 FOR i=1 TO 36
5510 READ r$(i)
5520 NEXT i
5530 FOR i=1 TO 12
5540 READ v(i)
5550 NEXT i
5560 FOR i=1 TO 6
5570 READ p(i,1): LET p(i,2)=1
5575 READ p(i,3)
5580 NEXT i
5590 FOR i=1 TO 8
5600 READ d(i)
5610 NEXT i
5620 RETURN
5990 DATA 21,1,22,2,22,1,23,2,18
,1,19,2,16,3,20,4,11,1,12,2,7,1,
8,2,7,4,3,3,2,1
6000 DATA "oost","west","noord",
"zuid","boven","beneden","doden",
"help","meenemen","neerleggen",
"status","verband","lantaren aan",
"lantaren uit","lezen"
6010 DATA "de lantaren","de zak
meel","de waterzak","het kookboe
k","het pamflet","het verband","
het zwaard","het koekie"
6020 DATA "het lichaam van I.T.",
"doed monster","de gehakte slan
g","de dode hond"
6025 DATA "I.T.","het monster","
de slang","de hellehond","barbec
ue","bom"
6030 DATA "resetgrot","n-grot","
geheime gang","regelkamer","g-gr
ot","e-grot","p-grot","zwarte ka
mer","a-grot","holle grot","o-gr
ot","leegte"
6032 DATA "kleine grot","rotsgro
t","stankgrot","drakengrot","sla
ngengrot","gele grot","watergrot",
"glibbergrot","voedselgrot","e
indgrot","gekleurde grot","ijsgr
ot"

```

```
6034 DATA "open plek","bossen","  
bossen","bossen","bossen","bosse  
n","bossen","bossen","bossen","b  
ossen","bossen","bossen"  
6040 DATA 34,30,28,21,14,15,13,0  
,0,0,0,0  
6050 DATA 34,2,16,15,17,4,29,2,8  
,1,25,1  
6060 DATA 1,5,9,10,11,7,6,2
```

DE WOLF EN DE VIJF KLEINE GEITJES

Een sprookje van Grimm? Nee, het gaat hier om een spannend bordspel! Dit spel wordt gespeeld op een bord dat op het scherm getoond wordt. Als het spel begint zijn de geiten verdeeld over de onderste helft van het bord. Deze geiten zien er een beetje ongewoon uit: het zijn kleine vierkantjes die ieder een nummer bevatten. De wolf staat in de linkerbovenhoek van het bord.

Daar gaan we dan! U bent de wolf en de computer bestuurt de geiten. U heeft gewonnen als U drie geiten opgegeten heeft. De computer wint als een van deze verbazingwekkende geiten U heeft verorberd.

U kunt beginnen ... onderin het beeld zien we:

U MAG ... ZETTEN DOEN

Bijvoorbeeld

U MAG 2 ZETTEN DOEN

Het aantal zetten dat U mag doen ligt altijd tussen 1 en 3. De wolf, met andere woorden U, mag binnen een beurt zowel horizontaal als verticaal bewegen, echter nooit diagonaal. Elke stap kan gezet worden doormiddel van de cursorbesturingstoetsen. Als de laatste stap van Uw zet U op een vierkantje brengt waar ook een geit staat, dan is die geit van U. U mag nooit over een vierkantje gaan waar een geit staat.

Er zijn in totaal 5 geiten en deze kunnen rustig over elkaar heenspringen. Geiten kunnen slechts in één richting bewegen. Het aantal stappen dat zij mogen doen staat steeds in de vakken zelf aangegeven. Dit aantal ligt altijd tussen de 1 en 5.

Laten we eens een voorbeeld bezien. Stel dat bij een geit het cijfer 3 staat. Dit houdt in dat deze geit de volgende stappen mag doen:

- 3 stappen naar links, of
- 3 stappen naar rechts, of
- 3 stappen naar voren, of
- 3 stappen naar achteren

Zo'n geit kan dus niet bijvoorbeeld een stap voorwaarts en twee stappen naar rechts doen.

Als de wolf, kortom Uzelf, een zet heeft uitgevoerd toont de computer:

NU IS HET MIJN BEURT

Is de wolf weer aan zet is, ziet U

U MAG ... ZETTEN DOEN


Aan het eind van het spel toont de computer:

U HEEFT GEWONNEN

Of, en dat is niet zo leuk:

U HEEFT VERLOREN

```
10 REM DE WOLF EN DE 5 KLEINE
GEITJES
100 GO SUB 1000
110 REM ZET INVOEREN
120 PRINT AT 19,0;"U MAG ";bw;"
ZETTEN DOEN
130 LET t$=INKEY$
140 LET drw=(t$="6" AND rw<rbma
x)-(t$="7" AND rw>rbmin)
150 LET dkw=(t$="8" AND kw<kbma
x)-(t$="5" AND kw>kbmin)
160 IF (drw+dkw)=0 THEN GO TO 1
30
170 IF ((SCREEN$(rw+drw,kw+dkw
)<>" ") AND (bw<>1)) THEN GO TO
130
180 LET bw=bw-1
190 PRINT AT rw,kw; PAPER 4;" "
200 LET rw=rw+drw: LET kw=kw+dk
w
210 PRINT AT rw,kw;"U"
220 IF bw>0 THEN GO TO 120
230 FOR i=1 TO 5
240 IF rw<>p(i,1) THEN GO TO 28
0
250 IF kw<>p(i,2) THEN GO TO 28
0
255 IF p(i,3)=0 THEN GO TO 280
260 LET p(i,3)=0
270 LET beat=beat+1
```



```

280 NEXT i
290 LET bw=INT (RND*3)+1
300 IF beat=3 THEN PRINT "U HEE
FT GEWONNEN!": STOP
310 PRINT AT 19,0;"NU IS HET MI
JN BEURT
350 REM waarderig
355 LET max=-400
360 FOR i=1 TO 5
370 IF p(i,3)=0 THEN GO TO 490
390 LET ks=p(i,2)
400 LET rs=p(i,1)+p(i,3)
410 GO SUB 600
420 LET rs=p(i,1)-p(i,3)
430 GO SUB 600
440 LET rs=p(i,1)
450 LET ks=p(i,2)+p(i,3)
460 GO SUB 600
470 LET ks=p(i,2)-p(i,3)
480 GO SUB 600
490 NEXT i
500 REM beweeg 5
510 PRINT AT p(imax,1),p(imax,2
); PAPER 4;" "
520 LET p(imax,1)=rmax
530 LET p(imax,2)=kmax
540 LET p(imax,3)=INT (RND*5)+1
550 PRINT AT rmax,kmax;p(imax,3
)
570 IF max<400 THEN GO TO 110
580 PRINT AT 19,0;"      U HEEF
T VERLOREN!
590 STOP
600 REM status
610 LET stat=0

```

```

620 IF ((rs<rbmin) OR (rs>rbmax
)) THEN RETURN
630 IF ((ks<kbmin) OR (ks>kbmax
)) THEN RETURN
640 LET d=ABS (rw-rs)+ABS (kw-k
s.)
650 IF d=bw THEN RETURN
660 IF d=(bw-2) THEN LET stat=-
200
670 IF d=0 THEN LET stat=500: G
O TO 760
680 IF d<=5 THEN LET stat=stat+
35
690 IF rs=rw THEN LET stat=stat
+40
700 IF ks=kw THEN LET stat=stat
+40
710 IF ABS (p(i,1)-rw)+ABS (p(i
,2)-kw)=bw THEN LET stat=stat+80
720 FOR t=1 TO 5
730 IF rs<>p(t,1) THEN LET stat
=stat+10
740 IF ks<>p(t,2) THEN LET stat
=stat+10
750 NEXT t
760 IF stat+RND<max THEN RETURN

770 LET max=stat: LET imax=i
780 LET rmax=rs: LET kmax=ks
790 RETURN
1000 REM initialisatie
1010 DIM p(5,3)
1015 LET rbmin=5: LET rbmax=14
1020 LET bw=2: LET max=-400
1025 LET kbmin=10: LET kbmax=19
1030 LET rw=rbmin: LET kw=kbmin
1040 LET beat=0
1042 FOR i=0 TO 9
1045 PRINT AT rbmin+i,kbmin; PAP
ER 4; "
1048 NEXT i
1050 FOR i=1 TO 5
1060 LET p(i,1)=INT (RND*5)+rbmi
n+5
1070 LET p(i,2)=INT (RND*10)+kbm
in
1080 LET p(i,3)=INT (RND*5)+1
1090 PRINT AT p(i,1),p(i,2);p(i,
3)
1100 NEXT i
1110 PRINT AT rw,kw;"U"
1120 RETURN


```

STUITERBAL

Heeft U weleens gekeken naar zo'n apart plastic balletje dat je op vreemde wijze door de kamer kunt laten stuiten?

Welnu ... wij hebben hier weer zo'n vreemde stuitbal die dan nu op het beeldscherm van Uw computer zal rondspringen. Deze bal heeft bovendien nog een extra eigenaardigheid die gewone stuitballen niet bezitten. De weg die onze bal volgt wordt afgebeeld en blijft zichtbaar, hierdoor kunnen we nu eindelijk eens zien wat voor mooie regelmatige patronen ontstaan door zo'n balletje simpelweg rond te laten stuiten.

```
5 REM STUITERBAL
10 LET X=RND*250: LET Y=RND*15
0
20 LET VX=0: LET VY=0
30 LET AX=RND*50+5: LET AY=RND
*50+5
40 PLOT X,Y
50 LET VX=VX+(127-X)/AX: LET X
=X+VX
60 LET VY=VY+(95-Y)/AY: LET Y=
Y+VY
70 DRAW VX,VY
80 GO TO 50
```



FORMULE 1 - RACE


Dames en heren, eindelijk is het zover, de auto's staan klaar voor de start, de motoren ronken voor een wellicht uitermate spannende Grand Prix.

Met het wereldkampioenschap in het vooruitzicht zal geen van deze coureurs een ander doel hebben dan te winnen.

Het is schitterend weer en het circuit ligt er prachtig bij, droog, dus alle voorwaarden voor een zeer snelle race zijn aanwezig. Wie weet, misschien worden er zelfs records gebroken! De laatste controle wordt uitgevoerd. De motoren draaien nu op volle toeren ... het startsein wordt gegeven, daar gaan ze ... !!

Ziezo dat was de introductie van dit fraaie spel. Nu wat zakelijke informatie. Na het RUN commando toont de computer KLAAR? en een paar seconden later bent U al op weg. U kunt Uw auto besturen door middel van de cursorbesturingstoetsen. Tijdens de race moet U niet alleen goed oppassen voor andere auto's, maar ook trachten zoveel mogelijk over de rode stippen, die op de weg staan, te rijden. Dit levert U extra punten op.

```
10 REM FORMULE 1-RACE
100 REM hoofdprogramma
110 GO SUB 1000
120 PRINT AT 0,5;"sc=";sc;"
    time left=";ti
130 GO SUB 300
135 GO SUB 500
140 PRINT AT rc,cc;" "
150 PRINT AT 21,0: POKE 23692,-
1: PRINT
160 GO SUB 600
170 LET ti=ti-1
180 IF ti>=0 THEN GO TO 120
190 PRINT AT 0,5;"UW SCORE IS:
";sc
200 PRINT AT 1,5;"-----
-----"
210 INPUT "Nog een keer (j of n
)? ";y$
220 IF y$="j" THEN GO TO 110
230 STOP
```

```

300 REM baan
310 IF (xw+dw)<8 OR (xw+dw+ww)>
240 THEN LET dw=0: LET lw=0
320 LET lw=lw-1
330 PLOT xw,yw: DRAW dw,-8
340 PLOT xw+ww,yw: DRAW dw,-8
350 LET xw=xw+dw
360 IF lw>0 THEN RETURN
370 LET lw=INT (RND*4)+2
380 LET dw=INT (RND*17)-9
390 RETURN
500 REM obstakels
510 LET ga=RND*150+sc
520 IF ga<150 THEN RETURN
530 LET co=INT (RND*3+xw/8)+1
540 IF ga>225 THEN PRINT AT 20,
co; INK 3;"A": RETURN
550 PRINT AT 20,co; INK 5;"o"
560 RETURN
600 REM auto
610 LET t$=INKEY$
620 LET cc=cc+(t$="8" AND cc<30
)-(t$="5" AND cc>1)
630 LET c$=SCREEN$(rc,cc)
640 PRINT AT rc,cc;"B"
650 IF c$=" " THEN LET sc=sc+1:
RETURN
660 IF c$="o" THEN LET sc=sc+10
: BEEP ,1,6: RETURN
680 LET sc=sc-25: BEEP 1,1
690 RETURN
1000 REM init
1010 LET rc=10: LET cc=15

```

```

1020 LET xw=110: LET yw=16
1030 LET lw=8: LET dw=0
1040 LET sc=0: LET ww=40
1050 LET ti=200
1060 RESTORE : RANDOMIZE : CLS
1070 FOR k=0 TO 7
1080 READ c
1090 POKE USA "a"+k,c
1100 NEXT k
1110 FOR k=0 TO 7
1120 READ c
1130 POKE USA "b"+k,c
1140 NEXT k
1150 PLOT xw,rc*8: DRAW 0,-rc*8
1160 PLOT xw+ww,rc*8: DRAW 0,-rc
*8
1170 PRINT AT 0,5: FLASH 1;"KLAA
R?"
1180 FOR i=1 TO 6: BEEP .5,i: NE
XT i
1190 RETURN
1200 DATA 24,68,68,24,24,24,68,6
8
1210 DATA 68,68,24,24,24,68,68,2
4

```

DE MARKTKRAMER

Heeft U zich op de markt wel eens verbaasd over de snelheid waarmee een marktkoopman verschillende bedragen kan optellen?

Dit spel is weer typisch een gezinsspel waarbij we kunnen uitvinden wie de beste marktkramer is. Steeds zien we een tweetal getallen die we natuurlijk moeten optellen. De opgaven worden echter steeds lastiger en wel op grond van het volgende. Laten we de volgende optelling eens nader bestuderen:

$$\begin{array}{r} 75856 \\ + 37637 \\ \hline \end{array}$$

Normaal zouden we eerst de getallen in de rechter kolom optellen, dan die in de kolom links daarvan, enzovoort. Hier moeten we echter eerst het meest linkse getal invoeren, vervolgens het daarop volgende getal, enz. Wel dit nu is het wat het spel steeds moeilijker en tegelijkertijd uitdagender maakt.

De praktijk leert dat dit zo eenvoudige spel werkelijk verslavend kan worden als we een record trachten te breken.

Om U een beetje te ontmoedigen ... ons record is 8. D.w.z. we konden nog twee getallen van precies 8 cijfers correct optellen en hierna ging het mis.

Kunt U het beter?

```
100 REM DE MARKTKRAMER
110 DIM a$(10,20)
120 FOR i=1 TO 10
130 READ a$(i)
140 NEXT i
150 LET ts=FN t(): LET p=1
160 CLS
170 LET a=INT (RND*10↑p+1)
180 LET b=INT (RND*10↑p+1)
190 PRINT AT 5,12;a;AT 6,12;b;"
+";AT 7,12;"-----"
200 INPUT "Antwoord: ";c
210 PRINT AT 6,12;c
220 LET ti=FN t()-ts
230 IF a+b=c THEN GO TO 280
240 PRINT "Fout, het was: ";a+b
250 LET p=p-1
260 GO TO 300
280 LET p=p+1
```




```

290 IF ti<60 THEN PAUSE 50: GO
TO 150
300 IF p>10 THEN LET p=10
310 PRINT "Score: ";a$(p+1)
320 PRINT "Uw tijd : ";ti;" sec
onden"
330 INPUT "Nog een keer? (j of
n)";y$
340 IF y$="j" THEN GO TO 150
350 STOP
390 DEF FN t()=INT ((65536*PEEK
23674+256*PEEK 23673+PEEK 23672
)/60)
400 DATA "grote nul","kleuter",
"krantejongen","klerk","autoverk
oper","makelaar","marktkramer","
olienagenaar","millionair","schoo
lmeester"

```

DOORZAKKEN

Aan het begin van dit eenvoudige maar desalniettemin intrigerende spel zien we acht horizontale balken met een aantal gaten. Boven deze balken zien we acht stenen die door de gaten kunnen vallen.



Steen 2 is in ons voorbeeld al door een gat gevallen. Het doel van het spel is nu alle stenen uiteindelijk onderin te krijgen en wel door de balken op de juiste wijze te laten verschuiven. Om een bepaalde balk te laten verschuiven, geven we steeds een commando van de volgende vorm:

BRS

waarin B het nummer van de balk voorstelt (van boven naar beneden genummerd van 1 tot 8). R de richting waarin de balk bewogen moet worden aangeeft (L voor links en R voor rechts) en S het aantal stappen is dat we de balk willen laten opschuiven (maximaal 9).

Bijvoorbeeld:

3R2

wil zeggen: laat balk 3 2 posities naar rechts opschuiven. Dit alles lijkt eenvoudig, maar het zal toch moeilijk blijken om deze klus in zo weinig mogelijk beurten te klaren.

```

5 REM DOORZAKKEN
10 DIM h(17,9): DIM z$(3)
20 LET sc=0: LET br=0
30 REM maak balken
40 FOR x=1 TO 17
50 FOR y=2 TO 9
60 LET h(x,y)=1
70 NEXT y
80 NEXT x
90 REM 4 gaten per balk
100 FOR y=2 TO 9
110 FOR b=1 TO 4
120 LET x=INT (RND*16+1)
130 LET h(x,y)=0
140 NEXT b
150 NEXT y
160 REM zet 1-6 erboven
170 FOR b=1 TO 6
180 LET h(2*b,1)=b+100
190 NEXT b
200 GO SUB 1000
210 GO SUB 2000
220 GO SUB 1000
500 INPUT "balk, l/r, stappen: ";
z$
510 LET y=VAL z$(1)+1
520 IF (y<2) OR (y>9) THEN GO TO 500
530 LET d$=z$(2)
540 LET dx=VAL z$(3)
550 IF dx=0 THEN GO TO 500
560 IF (d$<>"l") AND (d$<>"r")
THEN GO TO 500
570 IF d$="l" THEN GO SUB 3000
580 IF d$="r" THEN GO SUB 4000
590 LET br=br+1
600 GO SUB 2000
610 GO SUB 1000
620 IF sc=8 THEN STOP
630 IF dx>1 THEN LET dx=dx-1: G
O TO 570
640 GO TO 500
1000 FOR c=1 TO 9
1010 IF c=1 THEN PRINT " ";
1020 IF c>1 THEN PRINT c-1; " ";
1030 FOR x=1 TO 17
1040 LET z=h(x,c)
1050 GO SUB 1500
1060 NEXT x
1070 PRINT
1080 NEXT c
1090 PRINT " score: ";sc;" kee
r: ";br
1100 PRINT
1110 RETURN
1500 IF z=0 THEN PRINT " ";
1510 IF z=1 THEN PRINT PAPER 0;"
";

```

```

1520 IF z>100 THEN PRINT CHR$ (z
-52);
1530 RETURN
2000 FOR c=9 TO 1 STEP -1
2010 FOR x=1 TO 17
2020 LET ty=c
2030 IF h(x,c)<100 THEN GO TO 21
00
2040 IF ty>=9 THEN GO TO 2130
2050 IF h(x,ty+1)<>0 THEN GO TO
2100
2060 LET h(x,ty+1)=h(x,ty)
2070 LET h(x,ty)=0
2080 LET ty=ty+1
2090 GO TO 2040

```



```

2100 NEXT x
2110 NEXT c
2120 RETURN
2130 LET sc=sc+1
2140 LET h(x,ty)=0
2150 GO TO 2100
3000 LET tm=h(1,y)
3010 FOR x=1 TO 16
3020 LET h(x,y)=h(x+1,y)
3030 NEXT x
3040 LET h(17,y)=tm
3050 RETURN
4000 LET tm=h(17,y)
4010 FOR x=17 TO 2 STEP -1
4020 LET h(x,y)=h(x-1,y)
4030 NEXT x
4040 LET h(1,y)=tm
4050 RETURN

```


UFO AANVAL

Het is opmerkelijk hoeveel fantasie van pas kan komen om bepaalde spelletjes te spelen. We zullen dat illustreren aan de hand van het nu volgende spelletje, dat we allereerst zo precies en dus zo saai mogelijk beschrijven.

Onderin het beeld bevindt zich een „ding” dat er ongeveer als volgt uitziet:



Welnu, er vallen steeds vierkantjes naar beneden. Het „ding” kunnen we door middel van de cursorbesturingstoetsen bewegen, dit om te voorkomen dat het „ding” geraakt wordt.

Oké, dit klinkt allemaal wat saai en daarom zullen we nu een fantasierijke beschrijving geven.


U bent kapitein van het beste ruimteschip in ons heelal. Als U majestueus de Melkweg doorzweeft voelt U zich zeer trots op dit prachtige schip. Maar helaas als U eindelijk Uw thuisbasis nadert, kortom als het einde van de reis in zicht is, slaat het noodlot toe.

UFO's vallen aan!!!

Hoe kunt U aan deze zeer gevaarlijke invasie uit de ruimte ontkomen? Hoe kunt U Uzelf en dat prachtige ruimteschip redden? Waarom probeert U niet Uw

A - O - P - M of Atoom-Ontsnappings-Proton-Motor?

Hoe meer vliegende schotels U kunt ontwijken, hoe meer punten U krijgt. Ziezo ... dit klinkt al wat beter of niet dan?! In elk geval is dit hier om een snel, eenvoudig en attractief spel en daar gaat het uiteindelijk om!



```

10 REM UFO ANNUAL!
20 GO SUB 200
30 FOR i=1 TO 4: PRINT AT B(i)
,c(i); " ";: NEXT i
40 LET v=v+(INKEY$="8" AND v<2
3)-(INKEY$="5" AND v>3)
50 PRINT AT 20,v; INK 6;v$;

```


```

60 FOR i=1 TO 4
70 IF b(i)<20 THEN GO TO 120
80 LET b(i)=4: LET c(i)=v-3+IN
T (RND*11)
90 LET p=p+1: PRINT AT 0,1;p;
100 IF p=25 THEN LET s=2
110 GO TO 130
120 LET b(i)=b(i)+s
130 PRINT AT b(i),c(i); INK i;s
$;
140 IF b(i)<20 THEN GO TO 160
150 IF (c(i)-v<5 AND c(i)-v>0)
THEN BEEP 2,20: INK 0: PAPER 7:
STOP
160 NEXT i
170 GO TO 30
199 REM initialisatie
200 PAPER 5: CLS
210 LET v$=" "+CHR$ 134+CHR$ 14
0+CHR$ 140+CHR$ 137+" "
220 LET s$=CHR$ 140
230 LET s=1: LET p=0: LET v=13
240 DIM b(4): DIM c(4)
250 FOR i=1 TO 4: READ b(i): RE
AD c(i): NEXT i
260 PRINT AT 0,0;" 0 PUNTEN";
270 RETURN
280 DATA 16,4,12,8,4,16,8,20

```


MINI MANCALA

Mini mancala is een originele creatie van C. Freelink. Het is gebaseerd op een oud Arabisch spel, waarbij stenen in kommetjes worden verdeeld. Er zijn vier bakjes: A en B zijn van de computer en C en D zijn van U. Als het spel begint liggen er twee stenen in elk bakje.



Om de beurt nemen de spelers de stenen uit één van hun eigen bakjes en verdelen ze één voor één tegen de klok in over de andere drie bakjes.

U kunt er bijvoorbeeld voor kiezen om de stenen uit bakje D te verdelen:



Na deze zet is bakje D leeg. Er zal trouwens altijd minstens één bakje leeg zijn want gedurende iedere zet mogen de te verdelen stenen nooit in het bakje waaruit ze gekomen zijn worden teruggelegd.

Om te winnen moeten alle stenen in een van Uw eigen bakjes liggen. Zo kan de tegenstander immers geen enkele zet meer doen. De bakjes worden in dit spel op het scherm voorgesteld door vierkantjes met getallen die het aantal stenen aangeven.


We kunnen allereerst een moeilijkheidsgraad kiezen (1, 2 of 3: 1 is het gemakkelijkst). Bovendien mogen we aangeven wie mag beginnen.

Als U aan de beurt bent vraagt de computer van welk bakje U de stenen wilt verspreiden. Uiteraard geeft de computer steeds aan wat zijn eigen zet is. Na iedere zet toont de computer hoe de stenen over de bakjes verdeeld zijn.

Het is in feite verbazingwekkend dat een dergelijk ingewikkeld spel met zo'n kort programma beschreven kan worden.

Tenslotte vestigen we de aandacht op het diagram dat hier is afgebeeld. Dit diagram toont alle mogelijke zetten, kortom alle strategische wegen die we in dit zo boeiende spel kunnen bewandelen. Het is een fraai voorbeeld van een met name strategisch spel.

```
10 REM *** MINI MANCALA ***
20 DIM m(2,2): DIM l(4,2)
30 GO TO 700
99 REM initialisatie ***
100 FOR i=1 TO 2
110 LET m(1,i)=2: LET m(2,i)=2
120 FOR j=1 TO 4
130 READ x: LET l(j,i)=x
140 NEXT j
150 NEXT i
160 LET turn=0: LET xturn=0: RE
TURN
170 DATA 1,2,2,1,1,1,2,2
199 REM schermlayout ***
200 CLS : PRINT AT 8,14;"2 2"
210 PRINT AT 10,14;"2 2"
212 PRINT AT 4,10;"MINI MANCALA
";
214 PRINT AT 7,12;"A      B";
215 PRINT AT 11,12;"C      D";
220 FOR i=0 TO 4 STEP 2
230 PLOT 108+i*8,115
240 DRAW 0,-32
250 PLOT 108,115-i*8
260 DRAW 32,0: NEXT i
265 RETURN
```



```

299 REM zet invoeren ***
300 PRINT AT 20,7;"NU IS HET UW
BEURT!";TAB 31;" ";
310 INPUT "WELK VAK? (C OF D)?"
";TAB 32;x$;
330 IF (x$="C" OR x$="c") AND m
(2,1)>0 THEN LET l=2: RETURN
340 IF (x$="D" OR x$="d") AND m
(2,2)>0 THEN LET l=3: RETURN
350 BEEP 1,3
360 GO TO 310
399 REM beurt computer ***
400 PRINT AT 20,7;"NU IS HET MI
JN BEURT!";
410 FOR w=0 TO 300: NEXT w
420 LET q=m(1,1)*1000+m(1,2)*10
0+m(2,1)*10+m(2,2)
430 LET l=4

```

```

440 IF m(1,1)>0 AND (lev<3 AND
RND*lev<.4) OR m(1,2)=0 OR q=143
0 OR q=1340 OR q=6110 OR q=1160
THEN LET l=1
450 PRINT AT 21,1;"IK VERDEEL D
E STENEN VAN VAK ";CHR$(65+(l=4
))
460 RETURN
499 REM verdelen ***
500 LET g=l(1,1); LET h=l(1,2)
510 IF m(g,h)=0 THEN RETURN
520 LET l=l+1: IF l>4 THEN LET
l=l-4
530 LET i=l(1,1); LET j=l(1,2)
540 GO SUB 600: GO TO 510
599 REM een steen verplaatsen *
600 FOR w=1 TO 200: NEXT w
610 LET m(g,h)=m(g,h)-1: LET m(
i,j)=m(i,j)+1
620 PRINT AT 6+2*g,12+2*h;m(g,h
);
630 PRINT AT 6+2*i,12+2*j;m(i,j
);
640 RETURN
699 REM hoofdprogramma ***
700 GO SUB 100: GO SUB 200
710 INPUT "NIVEAU? (1-3): ";a$
720 IF a$<>"1" AND a$<>"2" AND
a$<>"3" THEN GO TO 710
730 LET lev=VAL a$: PRINT AT 14
,12;"NIVEAU=";lev
740 INPUT "WIE BEGINT? JIJ (J)
OF IK (I)?";TAB 32;a$
750 IF a$="J" OR a$="j" THEN GO
TO 800
760 IF a$<>"I" AND a$<>"i" THEN
GO TO 740
770 FOR w=0 TO 200: NEXT w: GO
SUB 300: GO SUB 500
780 LET turn=turn+1: IF m(1,1)+
m(1,2)=0 THEN PRINT AT 20,1;"U H
EEFT GEWONNEN IN ";turn*xturn;"
BEURTEN!";TAB 31;" ";TAB 31;" ";
: GO TO 830
790 IF turn=40 THEN GO SUB 900:
IF turn=40 THEN GO TO 830
800 FOR w=0 TO 200: NEXT w: GO
SUB 400: GO SUB 500
810 LET turn=turn+1: IF m(2,1)+
m(2,2)=0 THEN PRINT AT 20,6;"IK
HEB GEWONNEN IN ";turn*xturn;" B
EURTEN!";TAB 31;" ";TAB 31;" ";
: GO TO 830
820 IF turn=40 THEN GO SUB 900:
IF turn=40 THEN GO TO 830
825 GO TO 770
830 INPUT "NOG EEN KEER?";TAB 3
2;"J OF N : ";a$

```



```

840 IF a$="J" OR a$="j" THEN RU
N
850 IF a$="N" OR a$="n" THEN ST
OP
860 GO TO 830
900 INPUT "IK STEL REMISE VOOR,
AKKOORD? J OF N : ";a$
910 IF a$="N" OR a$="n" THEN LE
T xturn=xturn+turn: LET turn=0:
RETURN
920 IF a$="J" OR a$="j" THEN RE
TURN
930 GO TO 900


```

STOP IT

Dit is nu eens een typisch voorbeeld van een spel dat hoewel zeer kort toch zeer boeiend is.

Heeft U wel eens getest hoe snel U reageert? Niet ... dan biedt dit spel U de mogelijkheid om dat eens uit te testen. Hoe sneller U reageert, hoe meer punten U haalt.

Het scherm ziet er als volgt uit:



In het vierkantje zien we steeds een letter staan. Als dit vierkantje nu over het scherm beweegt (van links naar rechts), moet U trachten het te stoppen en wel door zo snel mogelijk op een willekeurige toets te drukken als het vierkantje zich bevindt in de kolom die dezelfde letter draagt. De computer vraagt iedere keer hoe vaak we het spel willen spelen en welke moeilijkheidsgraad U wenst (1 is het gemakkelijkst, 2 is moeilijker en 3 is het moeilijkst). Aan het einde van het spel zien we dan onze score.

```

1 REM STOP IT
5 LET t$="abcd ": LET tu=0: L
ET hi=0
10 INPUT "moeilijkheidsgraad (
1-3): ";di: LET d=10-3*di
20 INPUT "aantal beurten: ";nt
30 FOR c=1 TO 5
40 FOR r=0 TO 15: PRINT AT r,1
4+3*c: PAPER 0;" "; NEXT r
50 PRINT AT 0,15+3*c;t$(c): NE
XT c
60 LET c=INT (RND*4+1): LET b$
=t$(c)
70 LET tu=tu+1: LET r=2+2*INT
(RND*7): LET c1=0
80 PRINT AT r,c1;" ";b$: FOR i
=1 TO d: NEXT i
90 LET c1=c1+1: IF (c1<31) AND
(INKEY$="") THEN GO TO 80
100 IF (INT ((c1-15)/3)<>c) OR
(c1=17+3*c) THEN GO TO 120
110 BEEP 1,7: LET hi=hi+1: GO T
O 130
120 BEEP 1,4: LET d$=INKEY$
130 PRINT AT 20,0;"score : ";hi
;" van de ";tu;" goed"
140 IF tu<nt THEN GO TO 60

```

BAAB

Als we het hebben over een „thema con variatione” dan hebben we het over een melodie die, hoewel gebaseerd op een reeds bekende melodie toch zijn eigen kenmerk draagt. Dit spel kan inderdaad beschouwd worden als een „thema con variatione” omdat het in zekere zin lijkt op een ander spel in dit boek. Het heeft echter toch een heel eigen karakter.

U ziet steeds 16 velden waarop ogenschijnlijk willekeurig A's en B's zijn geplaatst.

B	A	B	A
A	A	A	B
A	B	A	B
B	B	A	B

Welnu, als U een van de velden aangeeft (op het beeldscherm wordt precies getoond hoe dit moet) dan zullen alle letters in zowel de horizontale als verticale rijen die door dit aangegeven veld lopen veranderen. Dus elke A wordt nu een B en omgekeerd.


Het is in ieder geval erg eenvoudig: we moeten uiteindelijk het volgende resultaat krijgen:

A	B	B	A
A	B	B	A
A	B	B	A
A	B	B	A

```

5 REM BAAB
10 DIM b(4,4)
20 FOR r=1 TO 4: FOR c=1 TO 4
30 LET b(r,c)=INT (RND*2)
40 NEXT c: NEXT r
50 REM **schermlayout**
60 PRINT AT 5,5;"NUMMERS";TAB
19;"BORD"
70 LET i=0
80 FOR r=7 TO 13 STEP 2
90 FOR k=3 TO 12 STEP 3
100 LET i=i+1
110 PRINT AT r,k;i
120 PLOT k*8-4,60: DRAW 0,64
130 PLOT 20,164-r*8: DRAW 96,0
140 NEXT k
150 NEXT r
160 PLOT 20,124: DRAW 96,0: DRA
W 0,-64
170 FOR r=7 TO 13 STEP 2
180 FOR c=16 TO 24 STEP 2

```



```

190 IF b((r-5)/2,(c-16)/2) THEN
PRINT AT r,c;"B": GO TO 210
200 PRINT AT r,c;"A"
210 PLOT c*8-4,60: DRAW 0,64\
220 PLOT 140,164-r*8: DRAW 64,0
230 NEXT c
240 NEXT r
250 PLOT 140,124: DRAW 64,0: DR
AW 0,-64
300 REM **zet invoeren**
310 INPUT "NUMMER: ";k
320 IF k<1 OR k>16 THEN GO TO 3
10
330 LET r=INT ((k-1)/4)+1
340 LET c=k-(r-1)*4
350 FOR i=1 TO 4
360 LET b(r,i)=NOT b(r,i)
370 LET b(i,c)=NOT b(i,c)
380 NEXT i
390 LET b(r,c)=NOT b(r,c)
400 GO TO 170

```

KLINKERS EN MEDEKLINKERS

Dit competitie spel is nu eens een spel dat door het hele gezin gespeeld kan worden. De computer geeft U zeven letters. Als hij vraagt:

Klinker (k) of medeklinker (m):?

Dan typt U een K voor een klinker en een M voor een medeklinker. Als U op deze wijze zeven letters heeft verkregen begint het werkelijke spel: de computer daagt U nu uit om een zo groot mogelijk woord van deze letters te maken. Elke letter mag slechts éénmaal gebruikt worden. Een letter kan gewist worden door op de SPACE toets te drukken. Er geldt een tijdslimiet die op een klok getoond wordt.

Het is verbazingwekkend hoe spannend en moeilijk zo'n eenvoudig spel kan zijn!

```
10 REM klinkers & medeklinkers
20 INPUT "Aantal spelers? ";np
30 DIM s(np): DIM d$(14)
40 RANDOMIZE
50 LET c$="aeiouy"
60 LET v$="bcdfghjklmnpqrstvwxyz"
70 DEF FN t()=INT ((65536*PEEK
23674+256*PEEK 23673+PEEK 23672
)/60)
80 GO TO 1000
100 REM letters kiezen
110 PRINT AT 4,4;"Beurt van spe
ler ";p
120 PRINT AT 6,0;"Klinker (k) o
f medeklinker (m)?"
130 FOR i=2 TO 14 STEP 2
140 LET t$=INKEY$
150 IF t$<>"k" AND t$<>"m" THEN
GO TO 140
160 IF INKEY$<>" " THEN GO TO 15
0
170 IF t$="k" THEN LET d$(i)=c$
(INT (RAND*6+1))
180 IF t$="m" THEN LET d$(i)=v$
(INT (RAND*21+1))
190 PRINT AT 9,i;d$(i)
```




```

200 NEXT i
210 RETURN
300 REM woord invoeren
310 PRINT AT 0,0;"Resterende ti
jd:45 seconden"
320 LET ts=45+FN t()
330 LET l=0
340 LET a$=""
350 LET t$=INKEY$
360 IF CODE t$<>32 OR NOT l THEN
N GO TO 400
370 PRINT AT 9,l;" "
380 LET l=l+1
390 LET a$=a$( TO l)
400 IF t$<"a" OR t$>"z" THEN GO
TO 440
410 LET l=l+1
420 PRINT AT 9,l;t$
430 LET a$=a$+t$
440 LET ti=ts-FN t()
450 PRINT AT 0,16;ti;" "
460 IF ti>0 THEN GO TO 350
470 RETURN
500 REM punten berekenen
510 FOR j=1 TO l
520 FOR i=2 TO 14
530 IF a$(j)=d$(i) THEN GO TO 5
60
540 NEXT i
550 GO TO 610
560 LET d$(i)=" "
570 NEXT j
580 PRINT AT 11,0;"Speler ";p;"
: ";l;" punten"
590 LET s(p)=s(p)+l
600 RETURN
610 PRINT AT 11,0;"Speler ";p;"
gebruikte verkeerde letters"
620 RETURN
1000 REM hoofdprogramma
1010 FOR p=1 TO np
1020 CLS
1030 GO SUB 100
1040 CLS

```

```

1040 CLS
1050 PRINT AT 4,0;"Maak hier een
woord van:"
1055 PRINT AT 6,5;d$
1060 GO SUB 300
1070 GO SUB 500
1080 PRINT AT 21,0;"Druk op een
willekeurige toets"
1090 IF INKEY$="" THEN GO TO 109
0
1100 NEXT p
1110 CLS
1120 FOR i=1 TO np
1130 PRINT "Speler ";i;": ";s(i)
;" punten"
1140 NEXT i
1150 PRINT AT 20,0;"Nog een rond
e? (j of n) "
1160 LET t$=INKEY$
1170 IF t$="j" THEN GO TO 1010
1180 IF t$="n" THEN STOP
1190 GO TO 1160

```

MAANFASE ASTROLOGIE

Dit programma is gebaseerd op een studie van de Nederlandse natuurkundige en astroloog Dr Ir J. van Slooten. Hij was onderzoeker op het beroemde Natlab van Philips en besteedde al zijn vrije tijd aan astrologie. Hij ontwikkelde een zeer interessante astrologische theorie n.l. dat de stand (fase) van de maan op het moment van geboorte een zeer belangrijke invloed op het nog te vormen karakter zal hebben. Na het bestuderen van het leven van honderden mensen kwam hij tot de volgende conclusie: (we citeren nu letterlijk uit zijn boek maan-phase psychologie; uitgave Veen, Amersfoort) „De maanfase schijnt uit te drukken hoe de persoon zich met zijn zieleleven instelt tegenover de totaliteit van zijn medemensen, d.w.z. de maatschappij, en welke rol hij speelt in deze maatschappij; in laatste instantie: zijn missie in de wereld.”

Het diagram toont de cyclus van de maanfasen. Er zijn drie hoofdgroepen te onderscheiden n.l. wassende, volle en afnemende maan. Daarom zijn er ook drie typen mensen die op basis hiervan gedefiniëerd zijn.

INDIVIDUALISTEN:

geboren onder de wassende maan. Ze houden ervan om zelfstandig te werken, hebben een sterke wil en worden niet ontmoedigd door ongemakken. We citeren weer: „De individualisten werken ieder voor zich (zij „zien de anderen niet goed in het donker”) -vandaar hun naam. Een sterke innerlijke zekerheid van groei helpt hen over teleurstelling en tegenwerking heen; dagelijkse materiële zorgen zijn niet spoedig in staat, de uitvoering te beletten van datgene wat zij zich in het hoofd hebben gezet.”

SOCIALISTEN:

geboren onder de volle maan. Ze houden ervan in gemeenschappen te leven. Sleutelwoorden voor deze mensen zijn „vergelijken”, „kiezen” en „combineren”. Samenwerking is een overheersende factor, maar er is ook rivaliteit en afgunst. We citeren weer: „... De samenwerking is hier het sterkst, maar ook concurrentie; de behaagzucht, maar ook de minachting; de bewondering maar ook de verguizing. Werpt niet juist het helderste licht de donkerste schaduwen? ...”

BEHERENDEN:

zij zijn geboren onder een afnemende maan; zij weten dat het volle licht eens weer terug zal keren. Ze zijn voorzichtig en zeker van zichzelf, ze houden ervan alles onder controle te hebben. Of (citaat), „... niet dat zij afkerig zijn van roem en eer, maar zij aanvaarden (en eisen) die meer voor de zaak die zij voorstaan dan voor hun eigen persoon. „Ars longa, vita brevis”.”

Deze drie typen kunnen weer verder worden onderverdeeld om zo in het totaal 12 categorieën op te leveren.

Voortrekker

Deze personen lopen altijd voorop. Hun doelgerichtheid kan hen soms stijfkoppig laten zijn. Enerzijds kunnen we stellen dat deze personen erg eenzijdig zijn anderzijds blijkt dat in praktijk ook een van hun sterke punten.



Samentrekker

Tegenover de voortrekker staat de samentrekker, zij houden ervan met anderen samen te zijn, maar ze voelen dat ze „more equal” zijn dan hun metgezellen. Ze zijn romantisch en houden erg van het gezinsleven. Citaat: „Doet zich het geval voor dat een „voortrekker” een groep om zich heen vormt, dan zullen dat doorgaans lieden zijn van belangrijk minder betekenis: volgelingen, supporters, die de afgezonderdheid van de „ster” eerder onderstrepen dan opheffen. De „samentrekker” echter zoekt zich te omringen met gelijkwaardigen, wil „primus inter pares” zijn en trekt zich terug uit een kring van „min-vermogenden”.”



Realist

Staan met beide benen op de grond. Ze denken dat alles verklaard kan worden op basis van fysische verschijnselen. We zouden ze de bij uitstek „nuchteren” van onze samenleving kunnen noemen. Wonderen zijn er om onmiddellijk aan de hand van fysische wetten verklaard te worden.

**Spiritualist**

Zijn eigenlijk „niet van deze aarde”. Het zijn bij uitstek de zoekers naar het onbekende. Bovendien verwachten zij een verrassingselement van dat onbekende dat zij als een wonder zullen verwelkomen.

**Artiest**

Artiesten veranderen materialen op hun geheel eigen wijze. Ze kunnen kunstwerken scheppen maar ook nieuwe produkten ontwerpen. Bij de artiest gaat het niet om het materiaal waar hij vanuit gaat maar de nieuwe vorm.



Apostel

Deze personen zijn zelf niet erg creatief maar ze dragen eerder de ideeën van anderen over de aarde. Citaat „... Hier is het niet een eigen vinding die -zij het naïef of onbeholpen- wordt uitgebeeld, maar een andermans idee, waarvoor de aandacht wordt gevraagd en met grote toewijding „reclame” wordt gemaakt.”



Sleuteldrager


Voelen zich verantwoordelijk voor hun medemensen, vooral voor de medemensen die anders vergeten zouden worden. Ze zijn dus behulpzaam en zorgen voor anderen. Typische voorbeelden zijn Albert Schweitzer en William Booth. Sleuteldragers kunnen ook ter verantwoording worden geroepen en zo in feite in dubbele betekenis „zorgdrager” worden.”



Gezagsdrager

Nemen eveneens verantwoordelijkheid op zich maar dan met het doel het leven en gedrag van anderen te beïnvloeden. Ze kijken vooruit waar de meeste mensen liever terug zouden zien. Hun stijl wordt eerder door het woord „gebod” dan door het woord „verbod” gekarakteriseerd.





Ridder

Ridders zijn min of meer roekeloze avonturiers, altijd op zoek naar nieuwe ontdekkingen. Het zijn idealisten met een sterk eergevoel, „... De Ridder „zonder vrees of blaam” is de onverschrokken avonturier, de trotse kampioen, de heldhaftige strijder die de handschoen opneemt en met blijmoedige branie zijn leven waagt.” (citaat)



Arbiter

Deze lieden oordelen graag over andermans doen en laten. Ze voorstellen niet, maar als ze een beslissing nemen verwachten ze dat die ook geëerbiedigd wordt. Citaat: „Bij zijn uitspraak zal hij meestal een vorm kiezen die, hoe positief ook, toch niet kwetsend is en meer animeert dan irriteert.”



Revolutionair

Deze personen zijn meestal ontevreden met de wereld waarin wij leven. Ze hebben de neiging van alles alleen de slechte zijde te zien en willen de wereld veranderen. Nu zijn er helaas wel veel slechte zaken op onze aardbol aan te wijzen. De revolutionair zal deze aanwijzen en zal voor het oplossen ervan geen compromis willen aanvaarden.



Profeet

Profeten beschouwen onze wereld als het ware van bovenaf. Ze voorstellen de toekomst. Niet altijd echter wordt er naar hen geluisterd. Niet voor niets luidt het spreekwoord „ profeten worden niet in eigen land geëerd”.



Dit zeer korte programma vraagt alleen om Uw geboortedag. Op grond van dit gegeven wordt dan de maanfase berekend. Aan de hand van het getoonde schema kunnen we dan precies nagaan wat deze maanfase ons te zeggen heeft.

```

10 REM MAANFASE ASTROLOGIE
20 GO TO 1000
100 REM datum invoeren
110 CLS
135 PRINT
140 INPUT "Dag (dd): ";dd: PRIN
T "Dag : ";dd
150 INPUT "Maand (mm): ";m: PRI
NT "Maand: ";m
160 INPUT "Jaar (JJJJ): ";y: PR
INT "Jaar: ";y
170 IF m<3 THEN LET m=m+12: LET
y=y-1
180 RETURN
200 REM berekening
210 LET t=INT (365.25*y)+INT (3
0.6*(m+1))+dd-694036
220 LET t=t/36525
230 LET la=350.737486+1236*t*36
0
240 LET la=la+307*t+6*t/60
250 LET la=la+51.18*t/3600-5.17
*t*t/3600
260 LET la=la-INT ((la/360)*360
270 LET la=INT (la+.5)
280 RETURN
300 REM resultaat laten zien
310 PRINT : PRINT
320 PRINT "Maanfase=";la;" grad
en"
330 PRINT : PRINT "Druk op een
willekeurige toets"
340 IF INKEY$="" THEN GO TO 340
350 RETURN
1000 REM hoofdprogramma
1010 GO SUB 100: REM datum invoe
ren
1020 GO SUB 200: REM berekening
1030 GO SUB 300: REM resultaat l
aten zien
1040 GO TO 1010

```

Addison-Wesley Nederland

The following is a selection of titles from the Addison-Wesley Nederland micro-computing list:

Title code	Author	Title
10187	Birnbaum	How to Choose your Small Business Computer
14638	Boon	Explorer's Guide to the ZX Spectrum and ZX 81
13791	Bourne	The Unix System
10158	Brown	Pascal from BASIC
10242	Clark	Executive VisiCalc for the Apple
10243	Clark	Executive VisiCalc for the IBM PC
10113	Covvey	Computer Choices
01939	Covvey	Computer Consciousness
01589	Dwyer	BASIC and the Personal Computer
03115	Dwyer	A Bit of BASIC
14239	Ferguson	Assembly Language Programming on the BBC Micro
14468	Foley	Fundamentals of Interactive Computer Graphics
01995	Foster	Programming a Microcomputer
01937	Foster	Real Time Programming
03399	Gooze	The S6800 Family
06192	Greenberg	The Computer Image
14640	Hartnell	Games BBC Computers Play
14635	King	Programming the M68000
10341	Koffman	Pascal: A Problem-Solving Approach
05464	Lewis	Pascal for the IBM PC
04575	McCracken	A Guide to PL/M Programming
14059	McGregor	Advanced Programming Techniques for the BBC Micro
14058	McGregor	The BBC Micro Book
14514	McGregor	The Electron Book
05092	Myers	Microcomputer Graphics
05247	Nevison	The Little Book of BASIC Style: How to Write a Program You Can Read
05735	Press	Low Cost Word Processing
06272	Remer	Legal Care for Your Software
14637	Ross	Logo Programming
07773	Sargent	Interfacing Microcomputers to the Real World
06896	Sargent	The IBM PC from the Inside Out
14636	Sommerville	Information Unlimited
10286	Stanton	The Addison-Wesley Book of Atari Software 1983
07403	Stone	Microcomputer Interfacing
07768	Thornburg	Picture This!

07767	Thornburg	Picture This Too!
14691	Vincent	Problem Solving in BASIC (PET/VIC)
14678	Watt	Creating Adventure Programs on Your BBC Micro

These books are available through your local bookseller. For further information contact:

Addison-Wesley Nederland
De Lairessestraat 90
1071 PJ AMSTERDAM

Krijgt uw brein wel voldoende oefening? Is uw geest wel zo scherp en zijn uw reacties wel zo snel? Nee?

Masseer dan uw hersenen met deze geweldige verzameling spelen:

- boeiende avonturen (adventure games)
- wervelende akties (arcade games)
- denkspelen
- puzzels en kwissen
- behendigheidsspelen.

Al deze spelen zijn voor uw **ZX-SPECTRUM** volledig uitgewerkt. Deze kant-en-klaar programma's zijn als originele listings afgedrukt, zodat geen fouten kunnen optreden.

Dit boek bevat dertig spelen, waaronder:

- Zombies in het moeras
- Escher (hoe maak ik mijn eigen Escher?)
- Het avontuur met IT
- Galactische monsters
- New York, het fantastische spel tussen de wolkenkrabbers.

Treed nu de geweldige wereld van Renko & Edwards binnen!



ADDISON-WESLEY NEDERLAND



UITGEVERIJ OMIKRON

ISBN 0-201-16551-1

Renko & Edwards Spectaculaire spelen voor uw ZX-SPECTRUM

**Spectaculaire spelen
voor uw**

ZX-SPECTRUM

Hal Renko & Sam Edwards

